Rules for integrands involving exponentials of inverse tangents

1. $\int u\, e^{n \operatorname{ArcTan}[a x]}\, dx$

   1. $\int x^m\, e^{n \operatorname{ArcTan}[a x]}\, dx$

      1: $\int x^m\, e^{n \operatorname{ArcTan}[a x]}\, dx$ when $\frac{i n-1}{2} \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n \operatorname{ArcTan}[z]} == \frac{(1-i z)^{\frac{i n+1}{2}}}{(1+i z)^{\frac{i n-1}{2}} \sqrt{1+z^2}}$

Rule: If $\frac{i n-1}{2} \in \mathbb{Z}$, then

$$\int x^m\, e^{n \operatorname{ArcTan}[a x]}\, dx \longrightarrow \int x^m \frac{(1-i a x)^{\frac{i n+1}{2}}}{(1+i a x)^{\frac{i n-1}{2}} \sqrt{1+a^2 x^2}}\, dx$$

Program code:

```
Int[E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  Int[((1-I*a*x)^((I*n+1)/2)/((1+I*a*x)^((I*n-1)/2)*Sqrt[1+a^2*x^2])),x] /;
FreeQ[a,x] && IntegerQ[(I*n-1)/2]
```

```
Int[x_^m_.*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  Int[x^m*((1-I*a*x)^((I*n+1)/2)/((1+I*a*x)^((I*n-1)/2)*Sqrt[1+a^2*x^2])),x] /;
FreeQ[{a,m},x] && IntegerQ[(I*n-1)/2]
```

2: $\int x^m\, e^{n \operatorname{ArcTan}[a x]}\, dx$ when $\frac{i n-1}{2} \notin \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n \operatorname{ArcTan}[z]} == \frac{(1-i z)^{i n/2}}{(1+i z)^{i n/2}}$

Rule: If $\frac{i n-1}{2} \notin \mathbb{Z}$, then

$$\int x^m\, e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow \int x^m\,\frac{(1-i\,a\,x)^{\frac{i\,n}{2}}}{(1+i\,a\,x)^{\frac{i\,n}{2}}}\,dx$$

–

Program code:

```
Int[E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  Int[(1-I*a*x)^(I*n/2)/(1+I*a*x)^(I*n/2),x] /;
FreeQ[{a,n},x] && Not[IntegerQ[(I*n-1)/2]]
```

```
Int[x_^m_.*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  Int[x^m*(1-I*a*x)^(I*n/2)/(1+I*a*x)^(I*n/2),x] /;
FreeQ[{a,m,n},x] && Not[IntegerQ[(I*n-1)/2]]
```

2. $\int u\,(c+d\,x)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $a^2 c^2+d^2==0$

1: $\int u\,(c+d\,x)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $a^2 c^2+d^2==0\land(p\in\mathbb{Z}\lor c>0)$

Derivation: Algebraic simplification

Basis: $\text{ArcTan}[z]==-i\,\text{ArcTanh}[i\,z]$

Basis: $e^{n\,\text{ArcTanh}[z]}==\frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Note: Since $a^2 c^2+d^2==0$, the factor $\left(1+\frac{d\,x}{c}\right)^p$ will combine with one of the factors $(1-i\,a\,x)^{\frac{i\,n}{2}}$ or $(1+i\,a\,x)^{-\frac{i\,n}{2}}$.

Rule: If $a^2 c^2+d^2==0\land(p\in\mathbb{Z}\lor c>0)$, then

$$\int u\,(c+d\,x)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx\longrightarrow c^p\int u\left(1+\frac{d\,x}{c}\right)^p\frac{(1-i\,a\,x)^{\frac{i\,n}{2}}}{(1+i\,a\,x)^{\frac{i\,n}{2}}}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_)^p_.*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  c^p*Int[u*(1+d*x/c)^p*(1-I*a*x)^(I*n/2)/(1+I*a*x)^(I*n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c^2+d^2,0] && (IntegerQ[p] || GtQ[c,0])
```

2: $\int u\,(c+d\,x)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $a^2 c^2+d^2==0\land\neg(p\in\mathbb{Z}\lor c>0)$

Derivation: Algebraic simplification

Basis: $\text{ArcTan}[z]==-i\,\text{ArcTanh}[i\,z]$

Basis: $e^{n\,\text{ArcTanh}[z]}==\frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Note: Since $a^2 c^2+d^2==0$, the factor $(c+d\,x)^p$ will combine with one of the factors $(1-i\,a\,x)^{\frac{i\,n}{2}}$ or $(1+i\,a\,x)^{-\frac{i\,n}{2}}$ after piecewise constant extraction.

– Rule: If $a^2 c^2 + d^2 == 0 \land \neg (p \in \mathbb{Z} \lor c > 0)$, then

$$\int u (c + d x)^p e^{n \operatorname{ArcTan}[a x]} dx \longrightarrow \int \frac{u (c + d x)^p (1 - i a x)^{\frac{i n}{2}}}{(1 + i a x)^{\frac{i n}{2}}} dx$$

– Program code:

```
Int[u_.*(c_+d_.*x_)^p_.*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  Int[u*(c+d*x)^p*(1-I*a*x)^(I*n/2)/(1+I*a*x)^(I*n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c^2+d^2,0] && Not[IntegerQ[p] || GtQ[c,0]]
```

3. $\int u \left(c + \frac{d}{x}\right)^p e^{n \operatorname{ArcTan}[a x]} dx$ when $c^2 + a^2 d^2 == 0$

1: $\int u \left(c + \frac{d}{x}\right)^p e^{n \operatorname{ArcTan}[a x]} dx$ when $c^2 + a^2 d^2 == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $p \in \mathbb{Z}$, then $\left(c + \frac{d}{x}\right)^p == \frac{d^p}{x^p} \left(1 + \frac{c x}{d}\right)^p$

Rule: If $c^2 + a^2 d^2 == 0 \land p \in \mathbb{Z}$, then

$$\int u \left(c + \frac{d}{x}\right)^p e^{n \operatorname{ArcTan}[a x]} dx \longrightarrow d^p \int \frac{u}{x^p} \left(1 + \frac{c x}{d}\right)^p e^{n \operatorname{ArcTan}[a x]} dx$$

– Program code:

```
Int[u_.*(c_+d_./x_)^p_.*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  d^p*Int[u/x^p*(1+c*x/d)^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[c^2+a^2*d^2,0] && IntegerQ[p]
```

2. $\int u \left(c+\frac{d}{x}\right)^p e^{n \text{ArcTan}[a x]} \, dx$ when $c^2+a^2 d^2==0 \wedge p \notin \mathbb{Z}$

   1. $\int u \left(c+\frac{d}{x}\right)^p e^{n \text{ArcTan}[a x]} \, dx$ when $c^2+a^2 d^2==0 \wedge p \notin \mathbb{Z} \wedge \frac{i n}{2} \in \mathbb{Z}$

      1: $\int u \left(c+\frac{d}{x}\right)^p e^{n \text{ArcTan}[a x]} \, dx$ when $c^2+a^2 d^2==0 \wedge p \notin \mathbb{Z} \wedge \frac{i n}{2} \in \mathbb{Z} \wedge c>0$

Derivation: Algebraic simplification

Basis: $\text{ArcTan}[z] == -i \, \text{ArcTanh}[i z]$

Basis: If $\frac{n}{2} \in \mathbb{Z}$, then $e^{n \text{ArcTanh}[z]} == \frac{(1+z)^{n/2}}{(1-z)^{n/2}} == (-1)^{n/2} \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Note: Since $c^2+a^2 d^2==0$, the factor $\left(1+\frac{d}{c x}\right)^p$ will combine with the factor $\left(1-\frac{1}{i a x}\right)^{\frac{i n}{2}}$ or $\left(1+\frac{1}{i a x}\right)^{-\frac{i n}{2}}$.

Rule: If $c^2+a^2 d^2==0 \wedge p \notin \mathbb{Z} \wedge \frac{i n}{2} \in \mathbb{Z} \wedge c>0$, then

$$\int u \left(c+\frac{d}{x}\right)^p e^{n \text{ArcTanh}[a x]} \, dx \longrightarrow \int u \left(c+\frac{d}{x}\right)^p e^{-i n \text{ArcTanh}[i a x]} \, dx \longrightarrow (-1)^{n/2} c^p \int u \left(1+\frac{d}{c x}\right)^p \frac{\left(1-\frac{1}{i a x}\right)^{\frac{i n}{2}}}{\left(1+\frac{1}{i a x}\right)^{\frac{i n}{2}}} \, dx$$

Program code:

```
Int[u_.*(c_+d_./x_)^p_*E^(n_*ArcTanh[a_.*x_]),x_Symbol] :=
  (-1)^(n/2)*c^p*Int[u*(1+d/(c*x))^p*(1-1/(I*a*x))^(I*n/2)/(1+1/(I*a*x))^(I*n/2),x] /;
FreeQ[{a,c,d,p},x] && EqQ[c^2+a^2*d^2,0] && Not[IntegerQ[p]] && IntegerQ[I*n/2] && GtQ[c,0]
```

2: $\int u \left(c+\frac{d}{x}\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx$ when $c^2+a^2 d^2==0 \land p \notin \mathbb{Z} \land \frac{i n}{2} \in \mathbb{Z} \land \neg (c>0)$

Derivation: Algebraic simplification

Basis: $\operatorname{ArcTan}[z] == -i \operatorname{ArcTanh}[i z]$

Basis: $e^{n \operatorname{ArcTanh}[z]} == \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Rule: If $c^2+a^2 d^2==0 \land p \notin \mathbb{Z} \land \frac{i n}{2} \in \mathbb{Z} \land \neg (c>0)$, then

$$\int u \left(c+\frac{d}{x}\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx \longrightarrow \int u \left(c+\frac{d}{x}\right)^p e^{-i n \operatorname{ArcTanh}[i a x]} \, dx \longrightarrow \int u \left(c+\frac{d}{x}\right)^p \frac{(1-i a x)^{\frac{i n}{2}}}{(1+i a x)^{\frac{i n}{2}}} \, dx$$

Program code:

```
Int[u_.*(c_+d_./x_)^p_*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  Int[u*(c+d/x)^p*(1-I*a*x)^(I*n/2)/(1+I*a*x)^(I*n/2),x] /;
FreeQ[{a,c,d,p},x] && EqQ[c^2+a^2*d^2,0] && Not[IntegerQ[p]] && IntegerQ[I*n/2] && Not[GtQ[c,0]]
```

2: $\int u \left(c + \frac{d}{x}\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx$ when $c^2 + a^2 d^2 == 0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^p \left(c+\frac{d}{x}\right)^p}{\left(1+\frac{c x}{d}\right)^p} == 0$

Rule: If $c^2 + a^2 d^2 == 0 \land p \notin \mathbb{Z}$, then

$$\int u \left(c + \frac{d}{x}\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx \longrightarrow \frac{x^p \left(c + \frac{d}{x}\right)^p}{\left(1 + \frac{c x}{d}\right)^p} \int \frac{u}{x^p} \left(1 + \frac{c x}{d}\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx$$

Program code:

```
Int[u_.*(c_+d_./x_)^p_*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  x^p*(c+d/x)^p/(1+c*x/d)^p*Int[u/x^p*(1+c*x/d)^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c^2+a^2*d^2,0] && Not[IntegerQ[p]]
```

4. $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2\,c$

   1. $\int (c+d\,x^2)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2\,c$

      1. $\int (c+d\,x^2)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2\,c \land p < -1 \land i\,n \notin \mathbb{Z}$

         1: $\int \frac{e^{n\,\text{ArcTan}[a\,x]}}{(c+d\,x^2)^{3/2}}\,dx$ when $d == a^2\,c \land i\,n \notin \mathbb{Z}$

Rule: If $d == a^2\,c \land i\,n \notin \mathbb{Z}$, then

$$\int \frac{e^{n\,\text{ArcTan}[a\,x]}}{(c+d\,x^2)^{3/2}}\,dx \longrightarrow \frac{(n+a\,x)\,e^{n\,\text{ArcTan}[a\,x]}}{a\,c\,(n^2+1)\,\sqrt{c+d\,x^2}}$$

Program code:

```
Int[E^(n_.*ArcTan[a_.*x_])/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  (n+a*x)*E^(n*ArcTan[a*x])/(a*c*(n^2+1)*Sqrt[c+d*x^2]) /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && Not[IntegerQ[I*n]]
```

2: $\int \left(c + d\,x^2\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2 c \land p < -1 \land i\,n \notin \mathbb{Z} \land n^2 + 4\,(p+1)^2 \neq 0$

Rule: If $d == a^2 c \land p < -1 \land i\,n \notin \mathbb{Z} \land n^2 + 4\,(p+1)^2 \neq 0$, then

$$\int \left(c + d\,x^2\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow \frac{(n - 2\,a\,(p+1)\,x)\,\left(c + d\,x^2\right)^{p+1} e^{n\,\text{ArcTan}[a\,x]}}{a\,c\,\left(n^2 + 4\,(p+1)^2\right)} + \frac{2\,(p+1)\,(2\,p+3)}{c\,\left(n^2 + 4\,(p+1)^2\right)} \int \left(c + d\,x^2\right)^{p+1} e^{n\,\text{ArcTan}[a\,x]}\,dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  (n-2*a*(p+1)*x)*(c+d*x^2)^(p+1)*E^(n*ArcTan[a*x])/(a*c*(n^2+4*(p+1)^2)) +
  2*(p+1)*(2*p+3)/(c*(n^2+4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && LtQ[p,-1] && Not[IntegerQ[I*n]] && NeQ[n^2+4*(p+1)^2,0] && IntegerQ[2*p]
```

2. $\int \left(c + d\,x^2\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2 c \land (p \in \mathbb{Z} \lor c > 0)$

1: $\int \frac{e^{n\,\text{ArcTan}[a\,x]}}{c + d\,x^2}\,dx$ when $d == a^2 c$

Rule: If $d == a^2 c$, then

$$\int \frac{e^{n\,\text{ArcTan}[a\,x]}}{c + d\,x^2}\,dx \longrightarrow \frac{e^{n\,\text{ArcTan}[a\,x]}}{a\,c\,n}$$

Program code:

```
Int[E^(n_.*ArcTan[a_.*x_])/(c_+d_.*x_^2),x_Symbol] :=
  E^(n*ArcTan[a*x])/(a*c*n) /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c]
```

2: $\int \left(c+d\,x^2\right)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d==a^2\,c\ \land\ p\in\mathbb{Z}\ \land\ \frac{i\,n+1}{2}\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTan}[z]}==\frac{(1-i\,z)^{i\,n}}{\left(1+z^2\right)^{\frac{i\,n}{2}}}$

Rule: If $d==a^2\,c\ \land\ p\in\mathbb{Z}\ \land\ \frac{i\,n+1}{2}\in\mathbb{Z}$, then

$$\int \left(c+d\,x^2\right)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx\ \longrightarrow\ c^p\int \left(1+a^2\,x^2\right)^p\,\frac{(1-i\,a\,x)^{i\,n}}{\left(1+a^2\,x^2\right)^{\frac{i\,n}{2}}}\,dx\ \longrightarrow\ c^p\int \left(1+a^2\,x^2\right)^{p-\frac{i\,n}{2}}\,(1-i\,a\,x)^{i\,n}\,dx$$

–

Program code:

```
Int[(c_+d_.*x_^2)^p_.*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  c^p*Int[(1+a^2*x^2)^(p-I*n/2)*(1-I*a*x)^(I*n),x] /;
FreeQ[{a,c,d,p},x] && EqQ[d,a^2*c] && IntegerQ[p] && IntegerQ[(I*n+1)/2] && Not[IntegerQ[p-I*n/2]]
```

3: $\int (c + d x^2)^p e^{n \operatorname{ArcTan}[a x]} \, dx$ when $d == a^2 c \land (p \in \mathbb{Z} \lor c > 0)$

Derivation: Algebraic simplification

Basis: If $d == a^2 c \land p \in \mathbb{Z}$, then $(c + d x^2)^p == c^p (1 - i a x)^p (1 + i a x)^p$

Basis: $e^{n \operatorname{ArcTan}[z]} == \frac{(1 - i z)^{i n/2}}{(1 + i z)^{i n/2}}$

Rule: If $d == a^2 c \land (p \in \mathbb{Z} \lor c > 0)$, then

$$\int (c + d x^2)^p e^{n \operatorname{ArcTan}[a x]} \, dx \longrightarrow c^p \int (1 - i a x)^p (1 + i a x)^p \frac{(1 - i a x)^{\frac{i n}{2}}}{(1 + i a x)^{\frac{i n}{2}}} \, dx \longrightarrow c^p \int (1 - i a x)^{p + \frac{i n}{2}} (1 + i a x)^{p - \frac{i n}{2}} \, dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_.*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  c^p*Int[(1-I*a*x)^(p+I*n/2)*(1+I*a*x)^(p-I*n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[d,a^2*c] && (IntegerQ[p] || GtQ[c,0])
```

3. $\int (c+d\,x^2)^p\, e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0)$

1. $\int (c+d\,x^2)^p\, e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i\,n}{2} \in \mathbb{Z}$

1: $\int (c+d\,x^2)^p\, e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i\,n}{2} \in \mathbb{Z}^+$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTan}[z]} == \frac{(1-i\,z)^{i\,n}}{(1+z^2)^{\frac{i\,n}{2}}}$

Basis: If $d == a^2 c \land \frac{i\,n}{2} \in \mathbb{Z}$, then $(1+a^2 x^2)^{-\frac{i\,n}{2}} == c^{\frac{i\,n}{2}}\,(c+d\,x^2)^{-\frac{i\,n}{2}}$

Rule: If $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i\,n}{2} \in \mathbb{Z}^+$, then

$$\int (c+d\,x^2)^p\, e^{n\,\text{ArcTan}[a\,x]}\,dx \to \int (c+d\,x^2)^p\, \frac{(1-i\,a\,x)^{i\,n}}{(1+a^2 x^2)^{\frac{i\,n}{2}}}\,dx \to c^{\frac{i\,n}{2}} \int (c+d\,x^2)^{p-\frac{i\,n}{2}}\,(1-i\,a\,x)^{i\,n}\,dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  c^(I*n/2)*Int[(c+d*x^2)^(p-I*n/2)*(1-I*a*x)^(I*n),x] /;
FreeQ[{a,c,d,p},x] && EqQ[d,a^2*c] && Not[IntegerQ[p] || GtQ[c,0]] && IGtQ[I*n/2,0]
```

2: $\int (c+d\,x^2)^p\, e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i\,n}{2} \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTan}[z]} == \frac{(1+z^2)^{\frac{i\,n}{2}}}{(1+i\,z)^{i\,n}}$

Basis: If $d == a^2 c \land \frac{i\,n}{2} \in \mathbb{Z}$, then $(1+a^2 x^2)^{\frac{i\,n}{2}} == \frac{1}{c^{\frac{i\,n}{2}}}\,(c+d\,x^2)^{\frac{i\,n}{2}}$

Rule: If $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i n}{2} \in \mathbb{Z}^-$, then

$$\int (c+d x^2)^p e^{n \operatorname{ArcTan}[a x]} \, dx \to \int (c+d x^2)^p \frac{(1+a^2 x^2)^{\frac{i n}{2}}}{(1+i a x)^{i n}} \, dx \to \frac{1}{c^{\frac{i n}{2}}} \int \frac{(c+d x^2)^{p+\frac{i n}{2}}}{(1+i a x)^{i n}} \, dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  1/c^(I*n/2)*Int[(c+d*x^2)^(p+I*n/2)/(1+I*a*x)^(I*n),x] /;
FreeQ[{a,c,d,p},x] && EqQ[d,a^2*c] && Not[IntegerQ[p] || GtQ[c,0]] && ILtQ[I*n/2,0]
```

2: $\int (c+d x^2)^p e^{n \operatorname{ArcTan}[a x]} \, dx$ when $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i n}{2} \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $d == a^2 c$, then $\partial_x \frac{(c+d x^2)^p}{(1+a^2 x^2)^p} == 0$

Rule: If $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i n}{2} \notin \mathbb{Z}$, then

$$\int (c+d x^2)^p e^{n \operatorname{ArcTan}[a x]} \, dx \to \frac{c^{\operatorname{IntPart}[p]} (c+d x^2)^{\operatorname{FracPart}[p]}}{(1+a^2 x^2)^{\operatorname{FracPart}[p]}} \int (1+a^2 x^2)^p e^{n \operatorname{ArcTan}[a x]} \, dx$$

Program code:

```
Int[(c_+d_.*x_^2)^p_*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  c^IntPart[p]*(c+d*x^2)^FracPart[p]/(1+a^2*x^2)^FracPart[p]*Int[(1+a^2*x^2)^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[d,a^2*c] && Not[IntegerQ[p] || GtQ[c,0]]
```

2. $\int x^m (c+d x^2)^p e^{n \operatorname{ArcTan}[a x]} \, dx$ when $d == a^2 c$

1. $\int x (c+d x^2)^p e^{n \operatorname{ArcTan}[a x]} \, dx$ when $d == a^2 c \land p < -1 \land i n \notin \mathbb{Z}$

1: $\int \frac{x\, e^{n \operatorname{ArcTan}[a x]}}{\left(c+d x^2\right)^{3/2}}\, dx$ when $d == a^2 c \land i n \notin \mathbb{Z}$

Rule: If $d == a^2 c \land i n \notin \mathbb{Z}$, then

$$\int \frac{x\, e^{n \operatorname{ArcTan}[a x]}}{\left(c+d x^2\right)^{3/2}}\, dx \longrightarrow -\frac{(1-a n x)\, e^{n \operatorname{ArcTan}[a x]}}{d \left(n^2+1\right) \sqrt{c+d x^2}}$$

Program code:

```
Int[x_*E^(n_.*ArcTan[a_.*x_])/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  -(1-a*n*x)*E^(n*ArcTan[a*x])/(d*(n^2+1)*Sqrt[c+d*x^2]) /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && Not[IntegerQ[I*n]]
```

2: $\int x \left(c+d x^2\right)^p e^{n \operatorname{ArcTan}[a x]}\, dx$ when $d == a^2 c \land p < -1 \land i n \notin \mathbb{Z}$

Derivation: Integration by parts

Basis: $\partial_x \frac{\left(c+d x^2\right)^{p+1}}{2 d (p+1)} == x \left(c+d x^2\right)^p$

Rule: If $d == a^2 c \land p < -1 \land i n \notin \mathbb{Z}$, then

$$\int x \left(c+d x^2\right)^p e^{n \operatorname{ArcTan}[a x]}\, dx \longrightarrow \frac{\left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcTan}[a x]}}{2 d (p+1)} - \frac{a c n}{2 d (p+1)} \int \left(c+d x^2\right)^p e^{n \operatorname{ArcTan}[a x]}\, dx$$

$$\longrightarrow \frac{(2 (p+1)+a n x) \left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcTan}[a x]}}{a^2 c \left(n^2+4 (p+1)^2\right)} - \frac{n (2 p+3)}{a c \left(n^2+4 (p+1)^2\right)} \int \left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcTan}[a x]}\, dx$$

Program code:

```
Int[x_*(c_+d_.*x_^2)^p_*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  (c+d*x^2)^(p+1)*E^(n*ArcTan[a*x])/(2*d*(p+1)) - a*c*n/(2*d*(p+1))*Int[(c+d*x^2)^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && LtQ[p,-1] && Not[IntegerQ[I*n]] && IntegerQ[2*p]
```

```
(* Int[x_*(c_+d_.*x_^2)^p_*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  (2*(p+1)+a*n*x)*(c+d*x^2)^(p+1)*E^(n*ArcTan[a*x])/(a^2*c*(n^2+4*(p+1)^2)) -
  n*(2*p+3)/(a*c*(n^2+4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && LtQ[p,-1] && NeQ[n^2+4*(p+1)^2,0] && Not[IntegerQ[I*n]] *)
```

2. $\int x^2 \left(c+d x^2\right)^p e^{n \operatorname{ArcTanh}[a x]} \, dx$ when $a^2 c+d==0 \wedge p<-1 \wedge n \notin \mathbb{Z}$

1: $\int x^2 \left(c+d x^2\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx$ when $d==a^2 c \wedge n^2-2 (p+1)==0 \wedge i n \notin \mathbb{Z}$

Rule: If $d==a^2 c \wedge n^2-2 (p+1)==0 \wedge i n \notin \mathbb{Z}$, then

$$\int x^2 \left(c+d x^2\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx \longrightarrow -\frac{(1-a n x) \left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcTan}[a x]}}{a d n \left(n^2+1\right)}$$

Program code:

```
Int[x_^2*(c_+d_.*x_^2)^p_.*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  -(1-a*n*x)*(c+d*x^2)^(p+1)*E^(n*ArcTan[a*x])/(a*d*n*(n^2+1)) /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && EqQ[n^2-2*(p+1),0] && Not[IntegerQ[I*n]]
```

2: $\int x^2 \left(c+d x^2\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx$ when $d==a^2 c \wedge p<-1 \wedge i n \notin \mathbb{Z} \wedge n^2+4 (p+1)^2 \neq 0$

Derivation: Algebraic expansion and ???

Basis: $x^2 \left(c+d x^2\right)^p == -\frac{c \left(c+d x^2\right)^p}{d}+\frac{\left(c+d x^2\right)^{p+1}}{d}$

Rule: If $d==a^2 c \wedge p<-1 \wedge i n \notin \mathbb{Z} \wedge n^2+4 (p+1)^2 \neq 0$, then

$$\int x^2 \left(c+d x^2\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx \longrightarrow -\frac{c}{d} \int \left(c+d x^2\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx+\frac{1}{d} \int \left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcTan}[a x]} \, dx$$

$$\longrightarrow -\frac{(n-2(p+1)\,a\,x)\left(c+d\,x^2\right)^{p+1} e^{n\,\text{ArcTan}[a\,x]}}{a\,d\left(n^2+4(p+1)^2\right)}+\frac{n^2-2(p+1)}{d\left(n^2+4(p+1)^2\right)}\int\left(c+d\,x^2\right)^{p+1} e^{n\,\text{ArcTan}[a\,x]}\,dx$$

Program code:

```
Int[x_^2*(c_+d_.*x_^2)^p_*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  -(n-2*(p+1)*a*x)*(c+d*x^2)^(p+1)*E^(n*ArcTan[a*x])/(a*d*(n^2+4*(p+1)^2)) +
  (n^2-2*(p+1))/(d*(n^2+4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && LtQ[p,-1] && Not[IntegerQ[I*n]] && NeQ[n^2+4*(p+1)^2,0] && IntegerQ[2*p]
```

3. $\int x^m\left(c+d\,x^2\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d==a^2 c \land (p\in\mathbb{Z} \lor c>0)$

1: $\int x^m\left(c+d\,x^2\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d==a^2 c \land (p\in\mathbb{Z} \lor c>0) \land \frac{i\,n+1}{2}\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTan}[z]}==\frac{(1-i\,z)^{i\,n}}{\left(1+z^2\right)^{\frac{i\,n}{2}}}$

Rule: If $d==a^2 c \land (p\in\mathbb{Z} \lor c>0) \land \frac{i\,n+1}{2}\in\mathbb{Z}$, then

$$\int x^m\left(c+d\,x^2\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow c^p\int x^m\left(1+a^2x^2\right)^p\frac{(1-i\,a\,x)^{i\,n}}{\left(1+a^2x^2\right)^{\frac{i\,n}{2}}}\,dx \longrightarrow c^p\int x^m\left(1+a^2x^2\right)^{p-\frac{i\,n}{2}}(1-i\,a\,x)^{i\,n}\,dx$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_.*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  c^p*Int[x^m*(1+a^2*x^2)^(p-I*n/2)*(1-I*a*x)^(I*n),x] /;
FreeQ[{a,c,d,m,p},x] && EqQ[d,a^2*c] && (IntegerQ[p] || GtQ[c,0]) && IntegerQ[(I*n+1)/2] && Not[IntegerQ[p-I*n/2]]
```

2: $\int x^m \left(c+d x^2\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx$ when $d == a^2 c \wedge (p \in \mathbb{Z} \vee c > 0)$

Derivation: Algebraic simplification

Basis: If $d == a^2 c \wedge p \in \mathbb{Z}$, then $\left(c+d x^2\right)^p == c^p (1 - i a x)^p (1 + i a x)^p$

Basis: $e^{n \operatorname{ArcTan}[z]} == \frac{(1-i z)^{i n/2}}{(1+i z)^{i n/2}}$

Rule: If $d == a^2 c \wedge (p \in \mathbb{Z} \vee c > 0)$, then

$$\int x^m \left(c+d x^2\right)^p e^{n \operatorname{ArcTan}[a x]} \, dx \longrightarrow c^p \int x^m (1-i a x)^p (1+i a x)^p \frac{(1-i a x)^{\frac{i n}{2}}}{(1+i a x)^{\frac{i n}{2}}} \, dx \longrightarrow c^p \int x^m (1-i a x)^{p+\frac{i n}{2}} (1+i a x)^{p-\frac{i n}{2}} \, dx$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_.*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  c^p*Int[x^m*(1-I*a*x)^(p+I*n/2)*(1+I*a*x)^(p-I*n/2),x] /;
FreeQ[{a,c,d,m,n,p},x] && EqQ[d,a^2*c] && (IntegerQ[p] || GtQ[c,0])
```

4. $\int x^m \left(c + d x^2\right)^p e^{n \text{ArcTan}[a x]} \, dx$ when $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0)$

1. $\int x^m \left(c + d x^2\right)^p e^{n \text{ArcTan}[a x]} \, dx$ when $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i n}{2} \in \mathbb{Z}$

1: $\int x^m \left(c + d x^2\right)^p e^{n \text{ArcTan}[a x]} \, dx$ when $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i n}{2} \in \mathbb{Z}^+$

Derivation: Algebraic simplification

Basis: $e^{n \text{ArcTan}[z]} == \frac{(1 - i z)^{i n}}{\left(1 + z^2\right)^{\frac{i n}{2}}}$

Basis: If $d == a^2 c \land \frac{i n}{2} \in \mathbb{Z}$, then $\left(1 + a^2 x^2\right)^{-\frac{i n}{2}} == c^{\frac{i n}{2}} \left(c + d x^2\right)^{-\frac{i n}{2}}$

Rule: If $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i n}{2} \in \mathbb{Z}^+$, then

$$\int x^m \left(c + d x^2\right)^p e^{n \text{ArcTan}[a x]} \, dx \to \int x^m \left(c + d x^2\right)^p \frac{(1 - i a x)^{i n}}{\left(1 + a^2 x^2\right)^{\frac{i n}{2}}} \, dx \to c^{\frac{i n}{2}} \int x^m \left(c + d x^2\right)^{p - \frac{i n}{2}} (1 - i a x)^{i n} \, dx$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  c^(I*n/2)*Int[x^m*(c+d*x^2)^(p-I*n/2)*(1-I*a*x)^(I*n),x] /;
FreeQ[{a,c,d,m,p},x] && EqQ[d,a^2*c] && Not[IntegerQ[p] || GtQ[c,0]] && IGtQ[I*n/2,0]
```

2: $\int x^m \left(c + d x^2\right)^p e^{n \text{ArcTan}[a x]} \, dx$ when $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i n}{2} \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: $e^{n \text{ArcTan}[z]} == \frac{\left(1 + z^2\right)^{\frac{i n}{2}}}{(1 + i z)^{i n}}$

Basis: If $d == a^2 c \land \frac{i n}{2} \in \mathbb{Z}$, then $\left(1 + a^2 x^2\right)^{\frac{i n}{2}} == \frac{1}{c^{\frac{i n}{2}}} \left(c + d x^2\right)^{\frac{i n}{2}}$

Rule: If $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i n}{2} \in \mathbb{Z}^-$, then

$$\int x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcTan}[a x]} dx \longrightarrow \int x^m \left(c + d x^2\right)^p \frac{\left(1 + a^2 x^2\right)^{\frac{i n}{2}}}{(1 + i a x)^{i n}} dx \longrightarrow \frac{1}{c^{\frac{i n}{2}}} \int \frac{x^m \left(c + d x^2\right)^{p + \frac{i n}{2}}}{(1 + i a x)^{i n}} dx$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  1/c^(I*n/2)*Int[x^m*(c+d*x^2)^(p+I*n/2)/(1+I*a*x)^(I*n),x] /;
FreeQ[{a,c,d,m,p},x] && EqQ[d,a^2*c] && Not[IntegerQ[p] || GtQ[c,0]] && ILtQ[I*n/2,0]
```

2: $\int x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcTan}[a x]} dx$ when $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i n}{2} \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $d == a^2 c$, then $\partial_x \frac{\left(c+d x^2\right)^p}{\left(1+a^2 x^2\right)^p} == 0$

Rule: If $d == a^2 c \land \neg (p \in \mathbb{Z} \lor c > 0) \land \frac{i n}{2} \notin \mathbb{Z}$, then

$$\int x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcTan}[a x]} dx \longrightarrow \frac{c^{\operatorname{IntPart}[p]} \left(c + d x^2\right)^{\operatorname{FracPart}[p]}}{\left(1 + a^2 x^2\right)^{\operatorname{FracPart}[p]}} \int x^m \left(1 + a^2 x^2\right)^p e^{n \operatorname{ArcTan}[a x]} dx$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  c^IntPart[p]*(c+d*x^2)^FracPart[p]/(1+a^2*x^2)^FracPart[p]*Int[x^m*(1+a^2*x^2)^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,m,n,p},x] && EqQ[d,a^2*c] && Not[IntegerQ[p] || GtQ[c,0]]
```

3. $\int u\left(c+d x^2\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2 c$

1: $\int u\left(c+d x^2\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2 c \land (p \in \mathbb{Z} \lor c > 0)$

Derivation: Algebraic simplification

Basis: $e^{n\,\text{ArcTan}[z]} == \frac{(1-i z)^{\frac{i n}{2}}}{(1+i z)^{\frac{i n}{2}}}$

Basis: $\left(1+z^2\right)^p == (1-i z)^p (1+i z)^p$

Rule: If $d == a^2 c \land (p \in \mathbb{Z} \lor c > 0)$, then

$$\int u\left(c+d x^2\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow c^p \int u\,(1-i a x)^p (1+i a x)^p \frac{(1-i a x)^{\frac{i n}{2}}}{(1+i a x)^{\frac{i n}{2}}}\,dx \longrightarrow c^p \int u\,(1-i a x)^{p+\frac{i n}{2}} (1+i a x)^{p-\frac{i n}{2}}\,dx$$

Program code:

```
Int[u_*(c_+d_.*x_^2)^p_.*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  c^p*Int[u*(1-I*a*x)^(p+I*n/2)*(1+I*a*x)^(p-I*n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[d,a^2*c] && (IntegerQ[p] || GtQ[c,0])
```

2. $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2\,c \land \neg\,(p \in \mathbb{Z} \lor c > 0)$

1: $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2\,c \land \neg\,(p \in \mathbb{Z} \lor c > 0) \land \frac{i\,n}{2} \in \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $d == a^2\,c$, then $\partial_x \frac{(c+d\,x^2)^p}{(1-i\,a\,x)^p\,(1+i\,a\,x)^p} == 0$

Basis: $e^{n\,\text{ArcTan}[z]} == \frac{(1-i\,z)^{\frac{i\,n}{2}}}{(1+i\,z)^{\frac{i\,n}{2}}}$

Rule: If $d == a^2\,c \land \neg\,(p \in \mathbb{Z} \lor c > 0) \land \frac{i\,n}{2} \in \mathbb{Z}$, then

$$\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow \frac{c^{\text{IntPart}[p]}\,(c+d\,x^2)^{\text{FracPart}[p]}}{(1-i\,a\,x)^{\text{FracPart}[p]}\,(1+i\,a\,x)^{\text{FracPart}[p]}} \int u\,(1-i\,a\,x)^{p+\frac{i\,n}{2}}\,(1+i\,a\,x)^{p-\frac{i\,n}{2}}\,dx$$

Program code:

```
Int[u_*(c_+d_.*x_^2)^p_.*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  c^IntPart[p]*(c+d*x^2)^FracPart[p]/((1-I*a*x)^FracPart[p]*(1+I*a*x)^FracPart[p])*
    Int[u*(1-I*a*x)^(p+I*n/2)*(1+I*a*x)^(p-I*n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[d,a^2*c] && (IntegerQ[p] || GtQ[c,0]) && IntegerQ[I*n/2]
```

2: $\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $d == a^2\,c \land \neg\,(p \in \mathbb{Z} \lor c > 0) \land \frac{i\,n}{2} \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $d == a^2\,c$, then $\partial_x \frac{(c+d\,x^2)^p}{(1+a^2\,x^2)^p} == 0$

Rule: If $d == a^2\,c \land \neg\,(p \in \mathbb{Z} \lor c > 0) \land \frac{i\,n}{2} \notin \mathbb{Z}$, then

$$\int u\,(c+d\,x^2)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow \frac{c^{\text{IntPart}[p]}\,(c+d\,x^2)^{\text{FracPart}[p]}}{(1+a^2\,x^2)^{\text{FracPart}[p]}}\int u\,(1+a^2\,x^2)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_^2)^p_*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  c^IntPart[p]*(c+d*x^2)^FracPart[p]/(1+a^2*x^2)^FracPart[p]*Int[u*(1+a^2*x^2)^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[d,a^2*c] && Not[IntegerQ[p] || GtQ[c,0]] && Not[IntegerQ[I*n/2]]
```

5. $\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $c == a^2\,d$

1: $\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $c == a^2\,d \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c == a^2\,d \land p \in \mathbb{Z}$, then $\left(c+\frac{d}{x^2}\right)^p == \frac{d^p}{x^{2\,p}}\,(1+a^2\,x^2)^p$

Rule: If $c == a^2\,d \land p \in \mathbb{Z}$, then

$$\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow d^p\int \frac{u}{x^{2\,p}}\,(1+a^2\,x^2)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_.*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  d^p*Int[u/x^(2*p)*(1+a^2*x^2)^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[c-a^2*d,0] && IntegerQ[p]
```

2. $\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $c == a^2 d \wedge p \notin \mathbb{Z}$

1. $\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $c == a^2 d \wedge p \notin \mathbb{Z} \wedge \frac{i\,n}{2} \in \mathbb{Z}$

1: $\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $c == a^2 d \wedge p \notin \mathbb{Z} \wedge \frac{i\,n}{2} \in \mathbb{Z} \wedge c > 0$

Derivation: Algebraic simplification

Basis: $\left(1+z^2\right)^p == (1-i\,z)^p\,(1+i\,z)^p$

Rule: If $c == a^2 d \wedge p \notin \mathbb{Z} \wedge \frac{i\,n}{2} \in \mathbb{Z} \wedge c > 0$, then

$$\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow c^p \int u\left(1+\frac{1}{a^2 x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow c^p \int u\left(1-\frac{i}{a\,x}\right)^p \left(1+\frac{i}{a\,x}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  c^p*Int[u*(1-I/(a*x))^p*(1+I/(a*x))^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,p},x] && EqQ[c-a^2*d,0] && Not[IntegerQ[p]] && IntegerQ[I*n/2] && GtQ[c,0]
```

2: $\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $c == a^2 d \land p \notin \mathbb{Z} \land \frac{i\,n}{2} \in \mathbb{Z} \land \neg (c > 0)$

Derivation: Piecewise constant extraction

Basis: If $c == a^2 d$, then $\partial_x \frac{x^{2p}\left(c+\frac{d}{x^2}\right)^p}{(1-i\,a\,x)^p\,(1+i\,a\,x)^p} == 0$

Rule: If $c == a^2 d \land p \notin \mathbb{Z} \land \frac{i\,n}{2} \in \mathbb{Z} \land \neg (c > 0)$, then

$$\int u\left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow \frac{x^{2p}\left(c+\frac{d}{x^2}\right)^p}{(1-i\,a\,x)^p\,(1+i\,a\,x)^p}\int \frac{u}{x^{2p}}\,(1-i\,a\,x)^p\,(1+i\,a\,x)^p\,e^{n\,\text{ArcTan}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_*E^(n_*ArcTan[a_.*x_]),x_Symbol] :=
  x^(2*p)*(c+d/x^2)^p/((1-I*a*x)^p*(1+I*a*x)^p)*Int[u/x^(2*p)*(1-I*a*x)^p*(1+I*a*x)^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c-a^2*d,0] && Not[IntegerQ[p]] && IntegerQ[I*n/2] && Not[GtQ[c,0]]
```

2: $\int u \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$ when $c == a^2\,d \land p \notin \mathbb{Z} \land \frac{i\,n}{2} \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c == a^2\,d$, then $\partial_x \frac{x^{2\,p}\left(c+\frac{d}{x^2}\right)^p}{\left(1+a^2\,x^2\right)^p} == 0$

Rule: If $c == a^2\,d \land p \notin \mathbb{Z} \land \frac{i\,n}{2} \notin \mathbb{Z}$, then

$$\int u \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx \longrightarrow \frac{x^{2\,p}\left(c+\frac{d}{x^2}\right)^p}{\left(1+a^2\,x^2\right)^p} \int \frac{u}{x^{2\,p}} \left(1+a^2\,x^2\right)^p e^{n\,\text{ArcTan}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_*E^(n_.*ArcTan[a_.*x_]),x_Symbol] :=
  x^(2*p)*(c+d/x^2)^p/(1+a^2*x^2)^p*Int[u/x^(2*p)*(1+a^2*x^2)^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c-a^2*d,0] && Not[IntegerQ[p]] && Not[IntegerQ[I*n/2]]
```

2. $\int u\, e^{n\,\text{ArcTan}[a+b\,x]}\, dx$

1: $\int e^{n\,\text{ArcTan}[c\,(a+b\,x)]}\, dx$

Derivation: Algebraic simplification

Basis: $\text{ArcTan}[z] == -i\,\text{ArcTanh}[i\,z]$

Basis: $e^{n\,\text{ArcTanh}[z]} == \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Note: The second step of this composite rule would be unnecessary if *Mathematica* did not gratuitously simplify **ArcTanh[i z]** to **i ArcTan[z].**

Rule:

$$\int e^{n\,\text{ArcTan}[c\,(a+b\,x)]}\, dx \longrightarrow \int e^{-i\,n\,\text{ArcTanh}[i\,c\,(a+b\,x)]}\, dx \longrightarrow \int \frac{(1-i\,a\,c-i\,b\,c\,x)^{\frac{i\,n}{2}}}{(1+i\,a\,c+i\,b\,c\,x)^{\frac{i\,n}{2}}}\, dx$$

Program code:

```
Int[E^(n_.*ArcTan[c_.*(a_+b_.*x_)]),x_Symbol] :=
  Int[(1-I*a*c-I*b*c*x)^(I*n/2)/(1+I*a*c+I*b*c*x)^(I*n/2),x] /;
FreeQ[{a,b,c,n},x]
```

2. $\int (d+e\,x)^m\, e^{n\,\text{ArcTan}[c\,(a+b\,x)]}\, dx$

1: $\int x^m\, e^{n\,\text{ArcTan}[c\,(a+b\,x)]}\, dx$ when $m \in \mathbb{Z}^- \land -1 < i\,n < 1$

Derivation: Algebraic simplification and integration by substitution

Basis: $\text{ArcTan}[z] == -i\,\text{ArcTanh}[i\,z]$

Basis: $e^{n\,\text{ArcTanh}[z]} == \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Basis: If $m \in \mathbb{Z} \land -1 < i\,n < 1$, then

$$x^m \frac{(1-i\,c\,(a+b\,x))^{\frac{i\,n}{2}}}{(1+i\,c\,(a+b\,x))^{\frac{i\,n}{2}}} == \frac{4}{i^m\,n\,b^{m+1}\,c^{m+1}}\,\text{Subst}\left[\frac{x^{\frac{2}{i\,n}}\left(1-i\,a\,c-(1+i\,a\,c)\,x^{\frac{2}{i\,n}}\right)^m}{\left(1+x^{\frac{2}{i\,n}}\right)^{m+2}},\ x,\ \frac{(1-i\,c\,(a+b\,x))^{\frac{i\,n}{2}}}{(1+i\,c\,(a+b\,x))^{\frac{i\,n}{2}}}\right]\partial_x \frac{(1-i\,c\,(a+b\,x))^{\frac{i\,n}{2}}}{(1+i\,c\,(a+b\,x))^{\frac{i\,n}{2}}}$$

Note: There should be an algebraic substitution rule that makes this rule redundant.

Rule: If $m \in \mathbb{Z}^- \land -1 < i\,n < 1$, then

$$\int x^m\, e^{n\,\text{ArcTan}[c\,(a+b\,x)]}\, dx \to \int x^m\, e^{-i\,n\,\text{ArcTanh}[i\,c\,(a+b\,x)]}\, dx$$

$$\to \int x^m \frac{(1-i\,c\,(a+b\,x))^{\frac{i\,n}{2}}}{(1+i\,c\,(a+b\,x))^{\frac{i\,n}{2}}}\, dx$$

$$\to \frac{4}{i^m\,n\,b^{m+1}\,c^{m+1}}\,\text{Subst}\left[\int \frac{x^{\frac{2}{i\,n}}\left(1-i\,a\,c-(1+i\,a\,c)\,x^{\frac{2}{i\,n}}\right)^m}{\left(1+x^{\frac{2}{i\,n}}\right)^{m+2}}\, dx,\ x,\ \frac{(1-i\,c\,(a+b\,x))^{\frac{i\,n}{2}}}{(1+i\,c\,(a+b\,x))^{\frac{i\,n}{2}}}\right]$$

Program code:

```
Int[x_^m_*E^(n_*ArcTan[c_.*(a_+b_.*x_)]),x_Symbol] :=
  4/(I^m*n*b^(m+1)*c^(m+1))*
    Subst[Int[x^(2/(I*n))*(1-I*a*c-(1+I*a*c)*x^(2/(I*n)))^m/(1+x^(2/(I*n)))^(m+2),x],x,
      (1-I*c*(a+b*x))^(I*n/2)/(1+I*c*(a+b*x))^(I*n/2)] /;
FreeQ[{a,b,c},x] && ILtQ[m,0] && LtQ[-1,I*n,1]
```

2: $\int (d+e\,x)^m\, e^{n\,\text{ArcTan}[c\,(a+b\,x)]}\, dx$

Derivation: Algebraic simplification

Basis: $\text{ArcTan}[z] == -i\,\text{ArcTanh}[i\,z]$

Basis: $e^{n\,\text{ArcTanh}[z]} == \frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Rule:

$$\int (d+e\,x)^m\, e^{n\,\text{ArcTan}[c\,(a+b\,x)]}\, dx \longrightarrow \int (d+e\,x)^m\, e^{-i\,n\,\text{ArcTanh}[i\,c\,(a+b\,x)]}\, dx \longrightarrow \int (d+e\,x)^m\, \frac{(1-i\,a\,c-i\,b\,c\,x)^{\frac{i\,n}{2}}}{(1+i\,a\,c+i\,b\,c\,x)^{\frac{i\,n}{2}}}\, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*E^(n_.*ArcTan[c_.*(a_+b_.*x_)]),x_Symbol] :=
  Int[(d+e*x)^m*(1-I*a*c-I*b*c*x)^(I*n/2)/(1+I*a*c+I*b*c*x)^(I*n/2),x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

3. $\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcTan}[a+b\,x]}\,dx$ when $b\,d==2\,a\,e\wedge b^2\,c-e\left(1+a^2\right)==0$

1: $\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcTan}[a+b\,x]}\,dx$ when $b\,d==2\,a\,e\wedge b^2\,c-e\left(1+a^2\right)==0\wedge\left(p\in\mathbb{Z}\vee\frac{c}{1+a^2}>0\right)$

Derivation: Algebraic simplification

Basis: If $b\,d==2\,a\,e\wedge b^2\,c-e\left(1+a^2\right)==0$, then $c+d\,x+e\,x^2==\frac{c}{1+a^2}\left(1+(a+b\,x)^2\right)$

Basis: $\left(1+z^2\right)^p==(1-i\,z)^p\,(1+i\,z)^p$

Basis: $\text{ArcTan}[z]==-i\,\text{ArcTanh}[i\,z]$

Basis: $e^{n\,\text{ArcTanh}[z]}==\frac{(1+z)^{n/2}}{(1-z)^{n/2}}$

Rule: If $b\,d==2\,a\,e\wedge b^2\,c-e\left(1+a^2\right)==0\wedge\left(p\in\mathbb{Z}\vee\frac{c}{1+a^2}>0\right)$, then

$$\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcTan}[a+b\,x]}\,dx\to\left(\frac{c}{1+a^2}\right)^p\int u\left(1+(a+b\,x)^2\right)^p e^{n\,\text{ArcTan}[a+b\,x]}\,dx$$

$$\to\left(\frac{c}{1+a^2}\right)^p\int u\,(1-i\,a-i\,b\,x)^p\,(1+i\,a+i\,b\,x)^p\,\frac{(1-i\,a-i\,b\,x)^{\frac{i\,n}{2}}}{(1+i\,a+i\,b\,x)^{\frac{i\,n}{2}}}\,dx$$

$$\to\left(\frac{c}{1+a^2}\right)^p\int u\,(1-i\,a-i\,b\,x)^{p+\frac{i\,n}{2}}\,(1+i\,a+i\,b\,x)^{p-\frac{i\,n}{2}}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_+e_.*x_^2)^p_.*E^(n_.*ArcTan[a_+b_.*x_]),x_Symbol] :=
  (c/(1+a^2))^p*Int[u*(1-I*a-I*b*x)^(p+I*n/2)*(1+I*a+I*b*x)^(p-I*n/2),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[b*d,2*a*e] && EqQ[b^2*c-e(1+a^2),0] && (IntegerQ[p] || GtQ[c/(1+a^2),0])
```

2: $\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcTan}[a+b\,x]}\,dx$ when $b\,d==2\,a\,e\wedge b^2\,c-e\left(1+a^2\right)==0\wedge\neg\left(p\in\mathbb{Z}\vee\frac{c}{1+a^2}>0\right)$

Derivation: Piecewise constant extraction

Basis: If $b\,d == 2\,a\,e \land b^2\,c - e\,\left(1 + a^2\right) == 0$, then $\partial_x \frac{\left(c+d\,x+e\,x^2\right)^p}{\left(1+a^2+2\,a\,b\,x+b^2\,x^2\right)^p} == 0$

Rule: If $b\,d == 2\,a\,e \land b^2\,c - e\,\left(1 + a^2\right) == 0 \land \neg\left(p \in \mathbb{Z} \lor \frac{c}{1+a^2} > 0\right)$, then

$$\int u\,\left(c + d\,x + e\,x^2\right)^p\,e^{n\,\text{ArcTan}[a+b\,x]}\,dx \longrightarrow \frac{\left(c + d\,x + e\,x^2\right)^p}{\left(1 + a^2 + 2\,a\,b\,x + b^2\,x^2\right)^p}\int u\,\left(1 + a^2 + 2\,a\,b\,x + b^2\,x^2\right)^p\,e^{n\,\text{ArcTan}[a+b\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_+e_.*x_^2)^p_.*E^(n_.*ArcTan[a_+b_.*x_]),x_Symbol] :=
  (c+d*x+e*x^2)^p/(1+a^2+2*a*b*x+b^2*x^2)^p*Int[u*(1+a^2+2*a*b*x+b^2*x^2)^p*E^(n*ArcTan[a*x]),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[b*d,2*a*e] && EqQ[b^2*c-e(1+a^2),0] && Not[IntegerQ[p] || GtQ[c/(1+a^2),0]]
```

3: $\int u\,e^{n\,\text{ArcTan}\left[\frac{c}{a+b\,x}\right]}\,dx$

Derivation: Algebraic simplification

Basis: $\text{ArcTan}[z] == \text{ArcCot}\left[\frac{1}{z}\right]$

Rule:

$$\int u\,e^{n\,\text{ArcTan}\left[\frac{c}{a+b\,x}\right]}\,dx \longrightarrow \int u\,e^{n\,\text{ArcCot}\left[\frac{a}{c}+\frac{b\,x}{c}\right]}\,dx$$

Program code:

```
Int[u_.*E^(n_.*ArcTan[c_./(a_.+b_.*x_)]),x_Symbol] :=
  Int[u*E^(n*ArcCot[a/c+b*x/c]),x] /;
FreeQ[{a,b,c,n},x]
```

Rules for integrands involving exponentials of inverse cotangents

1. $\int u\, e^{n\,\text{ArcCot}[a\,x]}\,dx$

1: $\int u\, e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $\frac{i\,n}{2}\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $\frac{i\,n}{2}\in\mathbb{Z}$, then $e^{n\,\text{ArcCot}[z]} == (-1)^{\frac{i\,n}{2}}\, e^{-n\,\text{ArcTan}[z]}$

Rule: If $\frac{i\,n}{2}\in\mathbb{Z}$, then

$$\int u\, e^{n\,\text{ArcCot}[a\,x]}\,dx \longrightarrow (-1)^{\frac{i\,n}{2}}\int u\, e^{-n\,\text{ArcTan}[z]}\,dx$$

Program code:

```
Int[u_.*E^(n_*ArcCot[a_.*x_]),x_Symbol] :=
  (-1)^(I*n/2)*Int[u*E^(-n*ArcTan[a*x]),x] /;
FreeQ[a,x] && IntegerQ[I*n/2]
```

2. $\int u\, e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $\frac{i\,n}{2}\notin\mathbb{Z}$

1. $\int x^m\, e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $\frac{i\,n}{2}\notin\mathbb{Z}$

1. $\int x^m\, e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $\frac{i\,n}{2}\notin\mathbb{Z}\ \land\ m\in\mathbb{Z}$

1: $\int x^m\, e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $\frac{i\,n-1}{2}\in\mathbb{Z}\ \land\ m\in\mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: $e^{n \operatorname{ArcCot}[z]} == \frac{\left(1-\frac{i}{z}\right)^{\frac{i n+1}{2}}}{\left(1+\frac{i}{z}\right)^{\frac{i n-1}{2}} \sqrt{1+\frac{1}{z^2}}}$

Basis: $F\left[\frac{1}{x}\right] == -\frac{F\left[\frac{1}{x}\right]}{\left(\frac{1}{x}\right)^2} \partial_x \frac{1}{x}$

– Rule: If $\frac{i n-1}{2} \in \mathbb{Z} \land m \in \mathbb{Z}$, then

$$\int x^m e^{n \operatorname{ArcCot}[a x]} \, dx \to \int \frac{\left(1-\frac{i}{a x}\right)^{\frac{i n+1}{2}}}{\left(\frac{1}{x}\right)^m \left(1+\frac{i}{a x}\right)^{\frac{i n-1}{2}} \sqrt{1+\frac{1}{a^2 x^2}}} \, dx \to -\text{Subst}\left[\int \frac{\left(1-\frac{i x}{a}\right)^{\frac{i n+1}{2}}}{x^{m+2} \left(1+\frac{i x}{a}\right)^{\frac{i n-1}{2}} \sqrt{1+\frac{x^2}{a^2}}} \, dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[E^(n_*ArcCot[a_.*x_]),x_Symbol] :=
  -Subst[Int[(1-I*x/a)^((I*n+1)/2)/(x^2*(1+I*x/a)^((I*n-1)/2)*Sqrt[1+x^2/a^2]),x],x,1/x] /;
FreeQ[a,x] && IntegerQ[(I*n-1)/2]
```

```
Int[x_^m_.*E^(n_*ArcCot[a_.*x_]),x_Symbol] :=
  -Subst[Int[(1-I*x/a)^((I*n+1)/2)/(x^(m+2)*(1+I*x/a)^((I*n-1)/2)*Sqrt[1+x^2/a^2]),x],x,1/x] /;
FreeQ[a,x] && IntegerQ[(I*n-1)/2] && IntegerQ[m]
```

2: $\int x^m e^{n \operatorname{ArcCot}[a x]} \, dx$ when $i n \notin \mathbb{Z} \land m \in \mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: $e^{n \operatorname{ArcCot}[z]} == \frac{\left(1-\frac{i}{z}\right)^{\frac{i n}{2}}}{\left(1+\frac{i}{z}\right)^{\frac{i n}{2}}}$

Basis: $F\left[\frac{1}{x}\right] == -\frac{F\left[\frac{1}{x}\right]}{\left(\frac{1}{x}\right)^2} \partial_x \frac{1}{x}$

– Rule: If $i n \notin \mathbb{Z} \land m \in \mathbb{Z}$, then

$$\int x^m e^{n\,\text{ArcCot}[a\,x]}\,dx \longrightarrow \int x^m e^{i\,n\,\text{ArcCoth}[i\,a\,x]}\,dx \longrightarrow \int \frac{\left(1-\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}{\left(\frac{1}{x}\right)^m \left(1+\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}\,dx \longrightarrow -\text{Subst}\left[\int \frac{\left(1-\frac{i\,x}{a}\right)^{\frac{i\,n}{2}}}{x^{m+2} \left(1+\frac{i\,x}{a}\right)^{\frac{i\,n}{2}}}\,dx,\ x,\ \frac{1}{x}\right]$$

–

Program code:

```
Int[E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -Subst[Int[(1-I*x/a)^(I*n/2)/(x^2*(1+I*x/a)^(I*n/2)),x],x,1/x] /;
FreeQ[{a,n},x] && Not[IntegerQ[I*n]]
```

```
Int[x_^m_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -Subst[Int[(1-I*x/a)^(n/2)/(x^(m+2)*(1+I*x/a)^(n/2)),x],x,1/x] /;
FreeQ[{a,n},x] && Not[IntegerQ[I*n]] && IntegerQ[m]
```

2. $\int x^m e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $\frac{i\,n}{2}\notin\mathbb{Z}\land m\notin\mathbb{Z}$

1: $\int x^m e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $\frac{i\,n-1}{2}\in\mathbb{Z}\land m\notin\mathbb{Z}$

Derivation: Algebraic simplification, piecewise constant extraction and integration by substitution!

Basis: $e^{n\,\text{ArcCot}[z]}==\frac{\left(1-\frac{i}{z}\right)^{\frac{i\,n+1}{2}}}{\left(1+\frac{i}{z}\right)^{\frac{i\,n-1}{2}}\sqrt{1+\frac{1}{z^2}}}$

Basis: $\partial_x\left(x^m\left(\frac{1}{x}\right)^m\right)==0$

Basis: $F\left[\frac{1}{x}\right]==-\frac{F\left[\frac{1}{x}\right]}{\left(\frac{1}{x}\right)^2}\,\partial_x\frac{1}{x}$

Rule: If $\frac{i\,n-1}{2}\in\mathbb{Z}\land m\notin\mathbb{Z}$, then

$$\int x^m e^{n\,\text{ArcCot}[a\,x]}\,dx\longrightarrow x^m\left(\frac{1}{x}\right)^m\int\frac{\left(1-\frac{i}{a\,x}\right)^{\frac{i\,n+1}{2}}}{\left(\frac{1}{x}\right)^m\left(1+\frac{i}{a\,x}\right)^{\frac{i\,n-1}{2}}\sqrt{1+\frac{1}{a^2\,x^2}}}\,dx\longrightarrow -x^m\left(\frac{1}{x}\right)^m\text{Subst}\left[\int\frac{\left(1-\frac{i\,x}{a}\right)^{\frac{i\,n+1}{2}}}{x^{m+2}\left(1+\frac{i\,x}{a}\right)^{\frac{i\,n-1}{2}}\sqrt{1+\frac{x^2}{a^2}}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[x_^m_*E^(n_*ArcCot[a_.*x_]),x_Symbol] :=
  -x^m*(1/x)^m*Subst[Int[(1-I*x/a)^((I*n+1)/2)/(x^(m+2)*(1+I*x/a)^((I*n-1)/2)*Sqrt[1+x^2/a^2]),x],x,1/x] /;
FreeQ[{a,m},x] && IntegerQ[(I*n-1)/2] && Not[IntegerQ[m]]
```

2: $\int x^m e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $\frac{i\,n}{2}\notin\mathbb{Z}\land m\notin\mathbb{Z}$

Derivation: Algebraic simplification, piecewise constant extraction and integration by substitution!

Basis: $e^{n\,\text{ArcCot}[z]} = \frac{\left(1-\frac{i}{z}\right)^{\frac{i n}{2}}}{\left(1+\frac{i}{z}\right)^{\frac{i n}{2}}}$

Basis: $\partial_x \left(x^m \left(\frac{1}{x}\right)^m\right) = 0$

Basis: $F\left[\frac{1}{x}\right] = -\frac{F\left[\frac{1}{x}\right]}{\left(\frac{1}{x}\right)^2}\,\partial_x \frac{1}{x}$

- Rule: If $\frac{i n}{2} \notin \mathbb{Z} \land m \notin \mathbb{Z}$, then

$$\int x^m e^{n\,\text{ArcCot}[a x]}\,dx \longrightarrow x^m \left(\frac{1}{x}\right)^m \int \frac{\left(1-\frac{i}{a x}\right)^{\frac{i n}{2}}}{\left(\frac{1}{x}\right)^m \left(1+\frac{i}{a x}\right)^{\frac{i n}{2}}}\,dx \longrightarrow -x^m \left(\frac{1}{x}\right)^m \text{Subst}\left[\int \frac{\left(1-\frac{i x}{a}\right)^{\frac{i n}{2}}}{x^{m+2}\left(1+\frac{i x}{a}\right)^{\frac{i n}{2}}}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -Subst[Int[(1-I*x/a)^(n/2)/(x^(m+2)*(1+I*x/a)^(n/2)),x],x,1/x] /;
FreeQ[{a,m,n},x] && Not[IntegerQ[I*n/2]] && Not[IntegerQ[m]]
```

2. $\int u\,(c+d\,x)^p\,e^{n\,\mathrm{ArcCot}[a\,x]}\,dx$ when $a^2\,c^2+d^2=0\ \wedge\ \frac{i\,n}{2}\notin\mathbb{Z}$

   1: $\int u\,(c+d\,x)^p\,e^{n\,\mathrm{ArcCot}[a\,x]}\,dx$ when $a^2\,c^2+d^2=0\ \wedge\ \frac{i\,n}{2}\notin\mathbb{Z}\ \wedge\ p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $p\in\mathbb{Z}$, then $(c+d\,x)^p=d^p\,x^p\left(1+\frac{c}{d\,x}\right)^p$

Rule: If $a^2\,c^2+d^2=0\ \wedge\ \frac{i\,n}{2}\notin\mathbb{Z}\ \wedge\ p\in\mathbb{Z}$, then

$$\int u\,(c+d\,x)^p\,e^{n\,\mathrm{ArcCot}[a\,x]}\,dx\ \longrightarrow\ d^p\int u\,x^p\left(1+\frac{c}{d\,x}\right)^p e^{n\,\mathrm{ArcCot}[a\,x]}\,dx$$

–

Program code:

```
Int[u_.*(c_+d_.*x_)^p_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  d^p*Int[u*x^p*(1+c/(d*x))^p*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[a^2*c^2+d^2,0] && Not[IntegerQ[I*n/2]] && IntegerQ[p]
```

2: $\int u\,(c+d\,x)^p\,e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $a^2\,c^2+d^2==0\,\wedge\,\frac{i\,n}{2}\notin\mathbb{Z}\,\wedge\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{(c+d\,x)^p}{x^p\left(1+\frac{c}{d\,x}\right)^p}==0$

Rule: If $a^2\,c^2+d^2==0\,\wedge\,\frac{i\,n}{2}\notin\mathbb{Z}\,\wedge\,p\notin\mathbb{Z}$, then

$$\int u\,(c+d\,x)^p\,e^{n\,\text{ArcCot}[a\,x]}\,dx\,\longrightarrow\,\frac{(c+d\,x)^p}{x^p\left(1+\frac{c}{d\,x}\right)^p}\int u\,x^p\left(1+\frac{c}{d\,x}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_)^p_*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  (c+d*x)^p/(x^p*(1+c/(d*x))^p)*Int[u*x^p*(1+c/(d*x))^p*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[a^2*c^2+d^2,0] && Not[IntegerQ[I*n/2]] && Not[IntegerQ[p]]
```

3. $\int u\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $c^2+a^2\,d^2==0\,\wedge\,\frac{i\,n}{2}\notin\mathbb{Z}$

1. $\int x^m\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $c^2+a^2\,d^2==0\,\wedge\,\frac{i\,n}{2}\notin\mathbb{Z}\,\wedge\,(p\in\mathbb{Z}\,\vee\,c>0)$

1: $\int x^m\left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $c^2+a^2\,d^2==0\,\wedge\,\frac{i\,n}{2}\notin\mathbb{Z}\,\wedge\,(p\in\mathbb{Z}\,\vee\,c>0)\,\wedge\,m\in\mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: $\text{ArcCot}[z]==i\,\text{ArcCoth}[i\,z]$

Basis: $e^{n\,\text{ArcCoth}[z]}==\frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $F\left[\frac{1}{x}\right] == -\frac{F\left[\frac{1}{x}\right]}{\left(\frac{1}{x}\right)^2}\,\partial_x \frac{1}{x}$

Note: Since $c^2 + a^2 d^2 == 0$, the factor $\left(1+\frac{d\,x}{c}\right)^p$ will combine with the factor $\left(1-\frac{i\,x}{a}\right)^{\frac{i\,n}{2}}$ or $\left(1+\frac{i\,x}{a}\right)^{-\frac{i\,n}{2}}$.

Rule: If $c^2 + a^2 d^2 == 0 \land \frac{i\,n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c > 0) \land m \in \mathbb{Z}$, then

$$\int x^m \left(c+\frac{d}{x}\right)^p e^{n\,\mathrm{ArcCot}[a\,x]}\,dx \longrightarrow c^p \int \frac{1}{\left(\frac{1}{x}\right)^m}\left(1+\frac{d}{c\,x}\right)^p \frac{\left(1-\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}{\left(1+\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}\,dx \longrightarrow -c^p\,\mathrm{Subst}\left[\int \frac{\left(1+\frac{d\,x}{c}\right)^p \left(1-\frac{i\,x}{a}\right)^{\frac{i\,n}{2}}}{x^{m+2}\left(1+\frac{i\,x}{a}\right)^{\frac{i\,n}{2}}}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[(c_+d_./x_)^p_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -c^p*Subst[Int[(1+d*x/c)^p*(1-I*x/a)^(I*n/2)/(x^2*(1+I*x/a)^(I*n/2)),x],x,1/x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c^2+a^2*d^2,0] && Not[IntegerQ[I*n/2]] && (IntegerQ[p] || GtQ[c,0])
```

```
Int[x_^m_.*(c_+d_./x_)^p_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -c^p*Subst[Int[(1+d*x/c)^p*(1-I*x/a)^(I*n/2)/(x^(m+2)*(1+I*x/a)^(I*n/2)),x],x,1/x] /;
FreeQ[{a,c,d,m,n,p},x] && EqQ[c^2+a^2*d^2,0] && Not[IntegerQ[I*n/2]] && (IntegerQ[p] || GtQ[c,0]) && IntegerQ[m]
```

2: $\int x^m \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $c^2+a^2 d^2==0 \land \frac{i\,n}{2}\notin\mathbb{Z} \land (p\in\mathbb{Z} \lor c>0) \land m\notin\mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: $\text{ArcCot}[z] == i\,\text{ArcCoth}[i\,z]$

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $\partial_x \left(x^m \left(\frac{1}{x}\right)^m\right) == 0$

Basis: $F\left[\frac{1}{x}\right] == -\frac{F\left[\frac{1}{x}\right]}{\left(\frac{1}{x}\right)^2}\,\partial_x \frac{1}{x}$

Note: Since $c^2+a^2 d^2==0$, the factor $\left(1+\frac{d\,x}{c}\right)^p$ will combine with the factor $\left(1-\frac{i\,x}{a}\right)^{\frac{i\,n}{2}}$ or $\left(1+\frac{i\,x}{a}\right)^{-\frac{i\,n}{2}}$.

Rule: If $c^2+a^2 d^2==0 \land \frac{i\,n}{2}\notin\mathbb{Z} \land (p\in\mathbb{Z} \lor c>0) \land m\notin\mathbb{Z}$, then

$$\int x^m \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx \to c^p x^m \left(\frac{1}{x}\right)^m \int \frac{1}{\left(\frac{1}{x}\right)^m}\left(1+\frac{d}{c\,x}\right)^p \frac{\left(1-\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}{\left(1+\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}\,dx \to -c^p x^m \left(\frac{1}{x}\right)^m \text{Subst}\left[\int \frac{\left(1+\frac{d\,x}{c}\right)^p \left(1-\frac{i\,x}{a}\right)^{\frac{i\,n}{2}}}{x^{m+2}\left(1+\frac{i\,x}{a}\right)^{\frac{i\,n}{2}}}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[(c_+d_./x_)^p_*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  (c+d/x)^p/(1+d/(c*x))^p*Int[(1+d/(c*x))^p*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c^2+a^2*d^2,0] && Not[IntegerQ[I*n/2]] && Not[IntegerQ[p] || GtQ[c,0]]
```

```
Int[x_^m_*(c_+d_./x_)^p_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -c^p*x^m*(1/x)^m*Subst[Int[(1+d*x/c)^p*(1-I*x/a)^(I*n/2)/(x^(m+2)*(1+I*x/a)^(I*n/2)),x],x,1/x] /;
FreeQ[{a,c,d,m,n,p},x] && EqQ[c^2+a^2*d^2,0] && Not[IntegerQ[I*n/2]] && (IntegerQ[p] || GtQ[c,0]) && Not[IntegerQ[m]]
```

2: $\int u \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $c^2+a^2 d^2 == 0 \land \frac{i\,n}{2} \notin \mathbb{Z} \land \neg\,(p \in \mathbb{Z} \lor c > 0)$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\left(c+\frac{d}{x}\right)^p}{\left(1+\frac{d}{c\,x}\right)^p} == 0$

Rule: If $c^2+a^2 d^2 == 0 \land \frac{i\,n}{2} \notin \mathbb{Z} \land \neg\,(p \in \mathbb{Z} \lor c > 0)$, then

$$\int u \left(c+\frac{d}{x}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx \longrightarrow \frac{\left(c+\frac{d}{x}\right)^p}{\left(1+\frac{d}{c\,x}\right)^p} \int u \left(1+\frac{d}{c\,x}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_)^p_*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  (c+d/x)^p/(1+d/(c*x))^p*Int[u*(1+d/(c*x))^p*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c^2+a^2*d^2,0] && Not[IntegerQ[I*n/2]] && Not[IntegerQ[p] || GtQ[c,0]]
```

4. $\int x^m \left(c + d x^2\right)^p e^{n \operatorname{ArcCot}[a x]} \, dx$ when $d == a^2 c \land \frac{i n}{2} \notin \mathbb{Z}$

   1. $\int \left(c + d x^2\right)^p e^{n \operatorname{ArcCot}[a x]} \, dx$ when $d == a^2 c \land p \le -1$

      1: $\int \frac{e^{n \operatorname{ArcCot}[a x]}}{c + d x^2} \, dx$ when $d == a^2 c$

Rule: If $d == a^2 c$, then

$$\int \frac{e^{n \operatorname{ArcCot}[a x]}}{c + d x^2} \, dx \longrightarrow -\frac{e^{n \operatorname{ArcCot}[a x]}}{a c n}$$

Program code:

```
Int[E^(n_.*ArcCot[a_.*x_])/(c_+d_.*x_^2),x_Symbol] :=
  -E^(n*ArcCot[a*x])/(a*c*n) /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c]
```

2: $\int \frac{e^{n \operatorname{ArcCot}[a x]}}{\left(c + d x^2\right)^{3/2}} \, dx$ when $d == a^2 c \land \frac{i n + 1}{2} \notin \mathbb{Z}$

Note: When $\frac{i n+1}{2} \in \mathbb{Z}$, it is better to transform integrand into algebraic form.

Rule: If $d == a^2 c \land \frac{i n+1}{2} \notin \mathbb{Z}$, then

$$\int \frac{e^{n \operatorname{ArcCot}[a x]}}{\left(c + d x^2\right)^{3/2}} \, dx \longrightarrow -\frac{(n - a x) \, e^{n \operatorname{ArcCot}[a x]}}{a c \left(n^2 + 1\right) \sqrt{c + d x^2}}$$

Program code:

```
Int[E^(n_.*ArcCot[a_.*x_])/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  -(n-a*x)*E^(n*ArcCot[a*x])/(a*c*(n^2+1)*Sqrt[c+d*x^2]) /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && Not[IntegerQ[(I*n-1)/2]]
```

3: $\int (c+d x^2)^p e^{n \text{ArcCot}[a x]} \, dx$ when $d == a^2 c \wedge p < -1 \wedge p \neq -\frac{3}{2} \wedge n^2 + 4 (p+1)^2 \neq 0 \wedge \neg \left(p \in \mathbb{Z} \wedge \frac{i n}{2} \in \mathbb{Z}\right) \wedge \neg \left(p \notin \mathbb{Z} \wedge \frac{i n - 1}{2} \in \mathbb{Z}\right)$

- Rule: If $d == a^2 c \wedge p < -1 \wedge p \neq -\frac{3}{2} \wedge n^2 + 4 (p+1)^2 \neq 0 \wedge \neg \left(p \in \mathbb{Z} \wedge \frac{i n}{2} \in \mathbb{Z}\right) \wedge \neg \left(p \notin \mathbb{Z} \wedge \frac{i n - 1}{2} \in \mathbb{Z}\right)$, then

$$\int (c+d x^2)^p e^{n \text{ArcCot}[a x]} \, dx \longrightarrow -\frac{(n+2 a (p+1) x) \left(c+d x^2\right)^{p+1} e^{n \text{ArcCot}[a x]}}{a c \left(n^2+4 (p+1)^2\right)} + \frac{2 (p+1) (2 p+3)}{c \left(n^2+4 (p+1)^2\right)} \int (c+d x^2)^{p+1} e^{n \text{ArcCot}[a x]} \, dx$$

- Program code:

```
Int[(c_+d_.*x_^2)^p_*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -(n+2*a*(p+1)*x)*(c+d*x^2)^(p+1)*E^(n*ArcCot[a*x])/(a*c*(n^2+4*(p+1)^2)) +
  2*(p+1)*(2*p+3)/(c*(n^2+4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && LtQ[p,-1] && NeQ[p,-3/2] && NeQ[n^2+4*(p+1)^2,0] &&
  Not[IntegerQ[p] && IntegerQ[I*n/2]] && Not[Not[IntegerQ[p]] && IntegerQ[(I*n-1)/2]]
```

2. $\int x^m (c+d x^2)^p e^{n \text{ArcCot}[a x]} \, dx$ when $d == a^2 c \wedge m \in \mathbb{Z} \wedge 0 \leq m \leq -2 (p+1)$

1. $\int x (c+d x^2)^p e^{n \text{ArcCot}[a x]} \, dx$ when $d == a^2 c \wedge p \leq -1$

1: $\int \frac{x e^{n \text{ArcCot}[a x]}}{(c+d x^2)^{3/2}} \, dx$ when $d == a^2 c \wedge \frac{i n + 1}{2} \notin \mathbb{Z}$

- Rule: If $d == a^2 c \wedge \frac{i n + 1}{2} \notin \mathbb{Z}$, then

$$\int \frac{x e^{n \text{ArcCot}[a x]}}{(c+d x^2)^{3/2}} \, dx \longrightarrow -\frac{(1 + a n x) e^{n \text{ArcCot}[a x]}}{a^2 c \left(n^2+1\right) \sqrt{c+d x^2}}$$

- Program code:

```
Int[x_*E^(n_.*ArcCot[a_.*x_])/(c_+d_.*x_^2)^(3/2),x_Symbol] :=
  -(1+a*n*x)*E^(n*ArcCot[a*x])/(a^2*c*(n^2+1)*Sqrt[c+d*x^2]) /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && Not[IntegerQ[(I*n-1)/2]]
```

2: $\int x\left(c+d x^2\right)^p e^{n\,\text{ArcCot}[a x]}\,dx$ when $d = a^2 c \land p \le -1 \land p \ne -\frac{3}{2} \land n^2+4(p+1)^2 \ne 0 \land \neg\left(p \in \mathbb{Z} \land \frac{i n}{2} \in \mathbb{Z}\right) \land \neg\left(p \notin \mathbb{Z} \land \frac{i n-1}{2} \in \mathbb{Z}\right)$

Rule: If $d = a^2 c \land p \le -2 \land n^2+4(p+1)^2 \ne 0 \land \neg\left(p \in \mathbb{Z} \land \frac{i n}{2} \in \mathbb{Z}\right) \land \neg\left(p \notin \mathbb{Z} \land \frac{i n-1}{2} \in \mathbb{Z}\right)$, then

$$\int x\left(c+d x^2\right)^p e^{n\,\text{ArcCot}[a x]}\,dx \longrightarrow \frac{(2(p+1)-a n x)\left(c+d x^2\right)^{p+1} e^{n\,\text{ArcCot}[a x]}}{a^2 c\left(n^2+4(p+1)^2\right)}+\frac{n(2p+3)}{a c\left(n^2+4(p+1)^2\right)}\int\left(c+d x^2\right)^{p+1} e^{n\,\text{ArcCot}[a x]}\,dx$$

Program code:

```
Int[x_*(c_+d_.*x_^2)^p_*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  (2*(p+1)-a*n*x)*(c+d*x^2)^(p+1)*E^(n*ArcCot[a*x])/(a^2*c*(n^2+4*(p+1)^2)) +
  n*(2*p+3)/(a*c*(n^2+4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && LeQ[p,-1] && NeQ[p,-3/2] && NeQ[n^2+4*(p+1)^2,0] &&
  Not[IntegerQ[p] && IntegerQ[I*n/2]] && Not[Not[IntegerQ[p]] && IntegerQ[(I*n-1)/2]]
```

2. $\int x^2\left(c+d x^2\right)^p e^{n\,\text{ArcCot}[a x]}\,dx$ when $d = a^2 c \land p \le -1$

1: $\int x^2\left(c+d x^2\right)^p e^{n\,\text{ArcCot}[a x]}\,dx$ when $d = a^2 c \land n^2-2(p+1) = 0 \land n^2+1 \ne 0$

Rule: If $d = a^2 c \land n^2-2(p+1) = 0 \land n^2+1 \ne 0$, then

$$\int x^2\left(c+d x^2\right)^p e^{n\,\text{ArcCot}[a x]}\,dx \longrightarrow \frac{(n+2(p+1)a x)\left(c+d x^2\right)^{p+1} e^{n\,\text{ArcCot}[a x]}}{a^3 c\, n^2\left(n^2+1\right)}$$

Program code:

```
Int[x_^2*(c_+d_.*x_^2)^p_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  (n+2*(p+1)*a*x)*(c+d*x^2)^(p+1)*E^(n*ArcCot[a*x])/(a^3*c*n^2*(n^2+1)) /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && EqQ[n^2-2*(p+1),0] && NeQ[n^2+1,0]
```

2: $\int x^2 \left(c+d x^2\right)^p e^{n \operatorname{ArcCot}[a x]} \, dx$ when $d == a^2 c \land p \le -1 \land n^2 - 2(p+1) \neq 0 \land n^2 + 4(p+1)^2 \neq 0 \land \neg \left(p \in \mathbb{Z} \land \frac{i n}{2} \in \mathbb{Z}\right) \land \neg \left(p \notin \mathbb{Z} \land \frac{i n+1}{2} \in \mathbb{Z}\right)$

Rule: If

$d == a^2 c \land p \le -1 \land n^2 - 2(p+1) \neq 0 \land n^2 + 4(p+1)^2 \neq 0 \land \neg \left(p \in \mathbb{Z} \land \frac{i n}{2} \in \mathbb{Z}\right) \land \neg \left(p \notin \mathbb{Z} \land \frac{i n+1}{2} \in \mathbb{Z}\right)$,

then

$$\int x^2 \left(c+d x^2\right)^p e^{n \operatorname{ArcCot}[a x]} \, dx \longrightarrow \frac{(n+2(p+1) a x) \left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcCot}[a x]}}{a^3 c \left(n^2+4(p+1)^2\right)} + \frac{n^2-2(p+1)}{a^2 c \left(n^2+4(p+1)^2\right)} \int \left(c+d x^2\right)^{p+1} e^{n \operatorname{ArcCot}[a x]} \, dx$$

Program code:

```
Int[x_^2*(c_+d_.*x_^2)^p_*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  (n+2*(p+1)*a*x)*(c+d*x^2)^(p+1)*E^(n*ArcCot[a*x])/(a^3*c*(n^2+4*(p+1)^2)) +
  (n^2-2*(p+1))/(a^2*c*(n^2+4*(p+1)^2))*Int[(c+d*x^2)^(p+1)*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && LeQ[p,-1] && NeQ[n^2-2*(p+1),0] && NeQ[n^2+4*(p+1)^2,0] &&
  Not[IntegerQ[p] && IntegerQ[I*n/2]] && Not[Not[IntegerQ[p]] && IntegerQ[(I*n-1)/2]]
```

3: $\int x^m \left(c+d x^2\right)^p e^{n \text{ArcCot}[a x]} \, dx$ when $d == a^2 c \land m \in \mathbb{Z} \land 3 \leq m \leq -2 (p+1) \land p \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $d == a^2 c \land m \in \mathbb{Z} \land p \in \mathbb{Z}$, then
$x^m \left(c+d x^2\right)^p e^{n \text{ArcCot}[a x]} == -\frac{c^p}{a^{m+1}} \frac{e^{n \text{ArcCot}[a x]} \text{Cot}[\text{ArcCot}[a x]]^{m+2 (p+1)}}{\text{Cos}[\text{ArcCot}[a x]]^{2 (p+1)}} \partial_x \text{ArcCot}[a x]$

Rule: If $d == a^2 c \land m \in \mathbb{Z} \land 3 \leq m \leq -2 (p+1) \land p \in \mathbb{Z}$, then

$$\int x^m \left(c+d x^2\right)^p e^{n \text{ArcCot}[a x]} \, dx \longrightarrow -\frac{c^p}{a^{m+1}} \text{Subst}\left[\int \frac{e^{n x} \text{Cot}[x]^{m+2 (p+1)}}{\text{Cos}[x]^{2 (p+1)}} \, dx, x, \text{ArcCot}[a x]\right]$$

Program code:

```
Int[x_^m_.*(c_+d_.*x_^2)^p_*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -c^p/a^(m+1)*Subst[Int[E^(n*x)*Cot[x]^(m+2*(p+1))/Cos[x]^(2*(p+1)),x],x,ArcCot[a*x]] /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && IntegerQ[m] && LeQ[3,m,-2(p+1)] && IntegerQ[p]
```

3. $\int u\left(c+d x^2\right)^p e^{n \operatorname{ArcCot}[a x]} dx$ when $d == a^2 c \land \frac{i n}{2} \notin \mathbb{Z}$

1: $\int u\left(c+d x^2\right)^p e^{n \operatorname{ArcCot}[a x]} dx$ when $d == a^2 c \land \frac{i n}{2} \notin \mathbb{Z} \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $d == a^2 c \land p \in \mathbb{Z}$, then $\left(c+d x^2\right)^p == d^p x^{2p}\left(1+\frac{1}{a^2 x^2}\right)^p$

Rule: If $d == a^2 c \land \frac{i n}{2} \notin \mathbb{Z} \land p \in \mathbb{Z}$, then

$$\int u\left(c+d x^2\right)^p e^{n \operatorname{ArcCot}[a x]} dx \longrightarrow d^p \int u\, x^{2p}\left(1+\frac{1}{a^2 x^2}\right)^p e^{n \operatorname{ArcCot}[a x]} dx$$

Program code:

```
Int[u_.*(c_+d_.*x_^2)^p_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  d^p*Int[u*x^(2*p)*(1+1/(a^2*x^2))^p*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,c,d,n},x] && EqQ[d,a^2*c] && Not[IntegerQ[I*n/2]] && IntegerQ[p]
```

2: $\int u\left(c+d x^2\right)^p e^{n\,\text{ArcCot}[a x]}\,dx$ when $d==a^2 c \wedge \frac{i n}{2} \notin \mathbb{Z} \wedge p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $d == a^2 c$, then $\partial_x \frac{\left(c+d x^2\right)^p}{x^{2 p}\left(1+\frac{1}{a^2 x^2}\right)^p} == 0$

Rule: If $d == a^2 c \wedge \frac{i n}{2} \notin \mathbb{Z} \wedge p \notin \mathbb{Z}$, then

$$\int u\left(c+d x^2\right)^p e^{n\,\text{ArcCot}[a x]}\,dx \longrightarrow \frac{\left(c+d x^2\right)^p}{x^{2 p}\left(1+\frac{1}{a^2 x^2}\right)^p} \int u\, x^{2 p}\left(1+\frac{1}{a^2 x^2}\right)^p e^{n\,\text{ArcCot}[a x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_^2)^p_*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  (c+d*x^2)^p/(x^(2*p)*(1+1/(a^2*x^2))^p)*Int[u*x^(2*p)*(1+1/(a^2*x^2))^p*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[d,a^2*c] && Not[IntegerQ[I*n/2]] && Not[IntegerQ[p]]
```

5. $\int u \left(c+\frac{d}{x^2}\right)^p e^{n\,\mathrm{ArcCot}[a\,x]}\,dx$ when $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z}$

   1. $\int u \left(c+\frac{d}{x^2}\right)^p e^{n\,\mathrm{ArcCot}[a\,x]}\,dx$ when $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c>0)$

      1: $\int u \left(c+\frac{d}{x^2}\right)^p e^{n\,\mathrm{ArcCot}[a\,x]}\,dx$ when $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c>0) \land \left(2p \mid p+\frac{i\,n}{2}\right) \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $\mathrm{ArcCot}[z] == i\,\mathrm{ArcCoth}[i\,z]$

Basis: $e^{n\,\mathrm{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $\left(1+z^2\right)^p \frac{(1-i\,z)^n}{(1+i\,z)^n} == (1-i\,z)^{p+n}\,(1+i\,z)^{p-n}$

Basis: If $p+n \in \mathbb{Z}$, then $\left(1-\frac{i}{z}\right)^{p+n} \left(1+\frac{i}{z}\right)^{p-n} == \frac{(-1+i\,z)^{p-n}\,(1+i\,z)^{p+n}}{(i\,z)^{2p}}$

Rule: If $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c>0) \land \left(2p \mid p+\frac{i\,n}{2}\right) \in \mathbb{Z}$, then

$$\int u \left(c+\frac{d}{x^2}\right)^p e^{n\,\mathrm{ArcCot}[a\,x]}\,dx \to c^p \int u \left(1+\frac{1}{a^2 x^2}\right)^p \frac{\left(1-\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}{\left(1+\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}\,dx$$

$$\to c^p \int u \left(1-\frac{i}{a\,x}\right)^{p+\frac{i\,n}{2}} \left(1+\frac{i}{a\,x}\right)^{p-\frac{i\,n}{2}}\,dx$$

$$\to \frac{c^p}{(i\,a)^{2p}} \int \frac{u}{x^{2p}}\,(-1+i\,a\,x)^{p-\frac{i\,n}{2}}\,(1+i\,a\,x)^{p+\frac{i\,n}{2}}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  c^p/(I*a)^(2*p)*Int[u/x^(2*p)*(-1+I*a*x)^(p-I*n/2)*(1+I*a*x)^(p+I*n/2),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c,a^2*d] && Not[IntegerQ[I*n/2]] && (IntegerQ[p] || GtQ[c,0]) && IntegersQ[2*p,p+I*n/2]
```

2. $\int x^m \left(c+\frac{d}{x^2}\right)^p e^{n\,\mathrm{ArcCot}[a\,x]}\,dx$ when $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c > 0) \land \neg \left(2p \mid p+\frac{i\,n}{2}\right) \in \mathbb{Z}$

1: $\int x^m \left(c+\frac{d}{x^2}\right)^p e^{n\,\mathrm{ArcCot}[a\,x]}\,dx$ when $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c > 0) \land \neg \left(2p \mid p+\frac{i\,n}{2}\right) \in \mathbb{Z} \land m \in \mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: $\mathrm{ArcCot}[z] == i\,\mathrm{ArcCoth}[i\,z]$

Basis: $e^{n\,\mathrm{ArcCot}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $\left(1+z^2\right)^p \frac{(1-i\,z)^n}{(1+i\,z)^n} == (1-i\,z)^{p+n}\,(1+i\,z)^{p-n}$

Basis: $F\left[\frac{1}{x}\right] == -\frac{F\left[\frac{1}{x}\right]}{\left(\frac{1}{x}\right)^2}\,\partial_x \frac{1}{x}$

Rule: If $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c > 0) \land \neg \left(2p \mid p+\frac{i\,n}{2}\right) \in \mathbb{Z} \land m \in \mathbb{Z}$, then

$$\int x^m \left(c+\frac{d}{x^2}\right)^p e^{n\,\mathrm{ArcCot}[a\,x]}\,dx \longrightarrow c^p \int x^m \left(1+\frac{1}{a^2 x^2}\right)^p \frac{\left(1-\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}{\left(1+\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}\,dx$$

$$\longrightarrow c^p \int \frac{1}{\left(\frac{1}{x}\right)^m} \left(1-\frac{i}{a\,x}\right)^{p+\frac{i\,n}{2}} \left(1+\frac{i}{a\,x}\right)^{p-\frac{i\,n}{2}}\,dx$$

$$\longrightarrow -c^p\,\mathrm{Subst}\left[\int \frac{\left(1-\frac{i\,x}{a}\right)^{p+\frac{i\,n}{2}} \left(1+\frac{i\,x}{a}\right)^{p-\frac{i\,n}{2}}}{x^{m+2}}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[(c_+d_./x_^2)^p_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -c^p*Subst[Int[(1-I*x/a)^(p+I*n/2)*(1+I*x/a)^(p-I*n/2)/x^2,x],x,1/x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c,a^2*d] && Not[IntegerQ[I*n/2]] && (IntegerQ[p] || GtQ[c,0]) && Not[IntegerQ[2*p] && IntegerQ[p+I*n/2]]
```

```
Int[x_^m_.*(c_+d_./x_^2)^p_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -c^p*Subst[Int[(1-I*x/a)^(p+I*n/2)*(1+I*x/a)^(p-I*n/2)/x^(m+2),x],x,1/x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c,a^2*d] && Not[IntegerQ[I*n/2]] && (IntegerQ[p] || GtQ[c,0]) && Not[IntegerQ[2*p] && IntegerQ[p+I*n/2]] &&
  IntegerQ[m]
```

2: $\int x^m \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c > 0) \land \neg \left(2\,p \,\middle|\, p+\frac{i\,n}{2}\right) \in \mathbb{Z} \land m \notin \mathbb{Z}$

Derivation: Algebraic simplification and integration by substitution

Basis: $\text{ArcCot}[z] == i\,\text{ArcCoth}[i\,z]$

Basis: $e^{n\,\text{ArcCot}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: $\left(1+z^2\right)^p \frac{(1-i\,z)^n}{(1+i\,z)^n} == (1-i\,z)^{p+n}\,(1+i\,z)^{p-n}$

Basis: $F\left[\frac{1}{x}\right] == -\frac{F\left[\frac{1}{x}\right]}{\left(\frac{1}{x}\right)^2}\,\partial_x \frac{1}{x}$

Rule: If $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z} \land (p \in \mathbb{Z} \lor c > 0) \land \neg \left(2\,p \,\middle|\, p+\frac{i\,n}{2}\right) \in \mathbb{Z} \land m \in \mathbb{Z}$, then

$$\int x^m \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx \to c^p \int x^m \left(1+\frac{1}{a^2 x^2}\right)^p \frac{\left(1-\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}{\left(1+\frac{i}{a\,x}\right)^{\frac{i\,n}{2}}}\,dx$$

$$\to c^p\,x^m \left(\frac{1}{x}\right)^m \int \frac{1}{\left(\frac{1}{x}\right)^m} \left(1-\frac{i}{a\,x}\right)^{p+\frac{i\,n}{2}} \left(1+\frac{i}{a\,x}\right)^{p-\frac{i\,n}{2}}\,dx$$

$$\to -c^p\,x^m \left(\frac{1}{x}\right)^m \text{Subst}\left[\int \frac{\left(1-\frac{i\,x}{a}\right)^{p+\frac{i\,n}{2}} \left(1+\frac{i\,x}{a}\right)^{p-\frac{i\,n}{2}}}{x^{m+2}}\,dx,\ x,\ \frac{1}{x}\right]$$

–

Program code:

```
Int[x_^m_*(c_+d_./x_^2)^p_.*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  -c^p*x^m*(1/x)^m*Subst[Int[(1-I*x/a)^(p+I*n/2)*(1+I*x/a)^(p-I*n/2)/x^(m+2),x],x,1/x] /;
FreeQ[{a,c,d,m,n,p},x] && EqQ[c,a^2*d] && Not[IntegerQ[I*n/2]] && (IntegerQ[p] || GtQ[c,0]) && Not[IntegerQ[2*p] && IntegerQ[p+I*n/2]] &&
  Not[IntegerQ[m]]
```

2: $\int u \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx$ when $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z} \land \neg(p \in \mathbb{Z} \lor c > 0)$

Derivation: Piecewise constant extraction

Basis: If $c == a^2 d$, then $\partial_x \frac{\left(c+\frac{d}{x^2}\right)^p}{\left(1+\frac{1}{a^2 x^2}\right)^p} == 0$

Rule: If $c == a^2 d \land \frac{i\,n}{2} \notin \mathbb{Z} \land \neg(p \in \mathbb{Z} \lor c > 0)$, then

$$\int u \left(c+\frac{d}{x^2}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx \longrightarrow \frac{\left(c+\frac{d}{x^2}\right)^p}{\left(1+\frac{1}{a^2 x^2}\right)^p} \int u \left(1+\frac{1}{a^2 x^2}\right)^p e^{n\,\text{ArcCot}[a\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_./x_^2)^p_*E^(n_.*ArcCot[a_.*x_]),x_Symbol] :=
  (c+d/x^2)^p/(1+1/(a^2*x^2))^p*Int[u*(1+1/(a^2*x^2))^p*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,c,d,n,p},x] && EqQ[c,a^2*d] && Not[IntegerQ[I*n/2]] && Not[IntegerQ[p] || GtQ[c,0]]
```

2. $\int u\, e^{n\,\text{ArcCot}[a+b\,x]}\, dx$

1: $\int u\, e^{n\,\text{ArcCot}[a+b\,x]}\, dx$ when $\frac{i\,n}{2} \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $\frac{i\,n}{2} \in \mathbb{Z}$, then $e^{n\,\text{ArcCot}[z]} == (-1)^{\frac{i\,n}{2}}\, e^{-n\,\text{ArcTan}[z]}$

Rule: If $\frac{i\,n}{2} \in \mathbb{Z}$, then

$$\int u\, e^{n\,\text{ArcCot}[c\,(a+b\,x)]}\, dx \longrightarrow (-1)^{\frac{i\,n}{2}} \int u\, e^{-n\,\text{ArcTan}[c\,(a+b\,x)]}\, dx$$

Program code:

```
Int[u_.*E^(n_*ArcCot[c_.*(a_+b_.*x_)]),x_Symbol] :=
  (-1)^(I*n/2)*Int[u*E^(-n*ArcTan[c*(a+b*x)]),x] /;
FreeQ[{a,b,c},x] && IntegerQ[I*n/2]
```

2. $\int u\, e^{n \operatorname{ArcCot}[a+b x]}\, dx$ when $\frac{i n}{2} \notin \mathbb{Z}$

1: $\int e^{n \operatorname{ArcCot}[c (a+b x)]}\, dx$ when $\frac{i n}{2} \notin \mathbb{Z}$

Derivation: Algebraic simplification and piecewise constant extraction

Basis: $\operatorname{ArcCot}[z] == i \operatorname{ArcCoth}[i z]$

Basis: $e^{n \operatorname{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}} == \frac{z^{n/2}\left(1+\frac{1}{z}\right)^{n/2}}{(-1+z)^{n/2}}$

Basis: $\partial_x \frac{f[x]^n \left(1+\frac{1}{f[x]}\right)^n}{(1+f[x])^n} == 0$

Rule: If $\frac{i n}{2} \notin \mathbb{Z}$, then

$$\int e^{n \operatorname{ArcCot}[c (a+b x)]}\, dx \longrightarrow \int \frac{(i c (a+b x))^{\frac{i n}{2}} \left(1+\frac{1}{i c (a+b x)}\right)^{\frac{i n}{2}}}{(-1+i c (a+b x))^{\frac{i n}{2}}}\, dx \longrightarrow \frac{(i c (a+b x))^{\frac{i n}{2}} \left(1+\frac{1}{i c (a+b x)}\right)^{\frac{i n}{2}}}{(1+i a c+i b c x)^{\frac{i n}{2}}} \int \frac{(1+i a c+i b c x)^{\frac{i n}{2}}}{(-1+i a c+i b c x)^{\frac{i n}{2}}}\, dx$$

Program code:

```
Int[E^(n_.*ArcCot[c_.*(a_+b_.*x_)]),x_Symbol] :=
  (I*c*(a+b*x))^(I*n/2)*(1+1/(I*c*(a+b*x)))^(I*n/2)/(1+I*a*c+I*b*c*x)^(I*n/2)*
    Int[(1+I*a*c+I*b*c*x)^(I*n/2)/(-1+I*a*c+I*b*c*x)^(I*n/2),x] /;
FreeQ[{a,b,c,n},x] && Not[IntegerQ[I*n/2]]
```

2. $\int (d+e\,x)^m\, e^{n\,\text{ArcCoth}[c\,(a+b\,x)]}\, dx$ when $\frac{i\,n}{2} \notin \mathbb{Z}$

1: $\int x^m\, e^{n\,\text{ArcCot}[c\,(a+b\,x)]}\, dx$ when $m \in \mathbb{Z}^- \land -1 < i\,n < 1$

Derivation: Algebraic simplification and integration by substitution

Basis: $\text{ArcCot}[z] == i\,\text{ArcCoth}[i\,z]$

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}$

Basis: If $m \in \mathbb{Z} \land -1 < i\,n < 1$, then

$$x^m \frac{\left(1+\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}{\left(1-\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}} == \frac{4}{i^m\,n\,b^{m+1}\,c^{m+1}}\,\text{Subst}\left[\frac{x^{\frac{2}{i\,n}}\left(1+i\,a\,c+(1-i\,a\,c)\,x^{\frac{2}{i\,n}}\right)^m}{\left(-1+x^{\frac{2}{i\,n}}\right)^{m+2}},\ x,\ \frac{\left(1+\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}{\left(1-\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}\right]\partial_x \frac{\left(1+\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}{\left(1-\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}$$

Note: There should be an algebraic substitution rule that makes this rule redundant.

–

Rule: If $m \in \mathbb{Z}^- \land -1 < i\,n < 1$, then

$$\int x^m\, e^{n\,\text{ArcCot}[c\,(a+b\,x)]}\, dx \longrightarrow \int x^m \frac{\left(1+\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}{\left(1-\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}\, dx$$

$$\longrightarrow \frac{4}{i^m\,n\,b^{m+1}\,c^{m+1}}\,\text{Subst}\left[\int \frac{x^{\frac{2}{i\,n}}\left(1+i\,a\,c+(1-i\,a\,c)\,x^{\frac{2}{i\,n}}\right)^m}{\left(-1+x^{\frac{2}{i\,n}}\right)^{m+2}}\, dx,\ x,\ \frac{\left(1+\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}{\left(1-\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}\right]$$

Program code:

```
Int[x_^m_*E^(n_*ArcCoth[c_.*(a_+b_.*x_)]),x_Symbol] :=
  4/(I^m*n*b^(m+1)*c^(m+1))*
    Subst[Int[x^(2/(I*n))*(1+I*a*c+(1-I*a*c)*x^(2/(I*n)))^m/(-1+x^(2/(I*n)))^(m+2),x],x,
      (1+1/(I*c*(a+b*x)))^(I*n/2)/(1-1/(I*c*(a+b*x)))^(I*n/2)] /;
FreeQ[{a,b,c},x] && ILtQ[m,0] && LtQ[-1,I*n,1]
```

2: $\int (d+e\,x)^m\, e^{n\,\text{ArcCot}[c\,(a+b\,x)]}\, dx$ when $\frac{i\,n}{2} \notin \mathbb{Z}$

Derivation: Algebraic simplification and piecewise constant extraction

Basis: $\text{ArcCot}[z] == i\,\text{ArcCoth}[i\,z]$

Basis: $e^{n\,\text{ArcCoth}[z]} == \frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}} == \frac{z^{n/2}\left(1+\frac{1}{z}\right)^{n/2}}{(-1+z)^{n/2}}$

Basis: $\partial_x \frac{f[x]^n\left(1+\frac{1}{f[x]}\right)^n}{(1+f[x])^n} == 0$

Rule: If $\frac{i\,n}{2} \notin \mathbb{Z}$, then

$$\int (d+e\,x)^m\, e^{n\,\text{ArcCot}[c\,(a+b\,x)]}\, dx \longrightarrow \int (d+e\,x)^m \frac{(i\,c\,(a+b\,x))^{\frac{i\,n}{2}}\left(1+\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}{(-1+i\,c\,(a+b\,x))^{\frac{i\,n}{2}}}\, dx$$

$$\longrightarrow \frac{(i\,c\,(a+b\,x))^{\frac{i\,n}{2}}\left(1+\frac{1}{i\,c\,(a+b\,x)}\right)^{\frac{i\,n}{2}}}{(1+i\,a\,c+i\,b\,c\,x)^{\frac{i\,n}{2}}} \int (d+e\,x)^m \frac{(1+i\,a\,c+i\,b\,c\,x)^{\frac{i\,n}{2}}}{(-1+i\,a\,c+i\,b\,c\,x)^{\frac{i\,n}{2}}}\, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*E^(n_.*ArcCoth[c_.*(a_+b_.*x_)]),x_Symbol] :=
  (I*c*(a+b*x))^(I*n/2)*(1+1/(I*c*(a+b*x)))^(I*n/2)/(1+I*a*c+I*b*c*x)^(I*n/2)*
    Int[(d+e*x)^m*(1+I*a*c+I*b*c*x)^(I*n/2)/(-1+I*a*c+I*b*c*x)^(I*n/2),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && Not[IntegerQ[I*n/2]]
```

3. $\int u\left(c+d x+e x^2\right)^p e^{n\,\text{ArcCot}[a+b x]}\,dx$ when $\frac{i n}{2}\notin\mathbb{Z}\wedge b d==2 a e\wedge b^2 c-e\left(1+a^2\right)==0$

1: $\int u\left(c+d x+e x^2\right)^p e^{n\,\text{ArcCot}[a+b x]}\,dx$ when $\frac{i n}{2}\notin\mathbb{Z}\wedge b d==2 a e\wedge b^2 c-e\left(1+a^2\right)==0\wedge\left(p\in\mathbb{Z}\vee\frac{c}{1+a^2}>0\right)$

Derivation: Algebraic simplification and piecewise constant extraction

Basis: If $b d==2 a e\wedge b^2 c-e\left(1+a^2\right)==0$, then $c+d x+e x^2==\frac{c}{1+a^2}\left(1+(a+b x)^2\right)$

Basis: $\text{ArcCot}[z]==i\,\text{ArcCoth}[i z]$

Basis: $e^{n\,\text{ArcCoth}[z]}==\frac{\left(1+\frac{1}{z}\right)^{n/2}}{\left(1-\frac{1}{z}\right)^{n/2}}==\frac{z^{n/2}\left(1+\frac{1}{z}\right)^{n/2}}{(-1+z)^{n/2}}$

Basis: $\partial_x\frac{f[x]^n\left(1+\frac{1}{f[x]}\right)^n}{(1+f[x])^n}==0$

Basis: $\partial_x\frac{(1-f[x])^n}{(-1+f[x])^n}==0$

Basis: $\left(1+z^2\right)^p==(1-i z)^p(1+i z)^p$

Basis: $\frac{z^n\left(1+\frac{1}{z}\right)^n}{(1+z)^n}==\left(\frac{z}{1+z}\right)^n\left(\frac{1+z}{z}\right)^n$

Rule: If $\frac{i n}{2}\notin\mathbb{Z}\wedge b d==2 a e\wedge b^2 c-e\left(1+a^2\right)==0\wedge\left(p\in\mathbb{Z}\vee\frac{c}{1+a^2}>0\right)$, then

$$\int u\left(c+d x+e x^2\right)^p e^{n\,\text{ArcCot}[a+b x]}\,dx\to\left(\frac{c}{1+a^2}\right)^p\int u\left(1+(a+b x)^2\right)^p\frac{(i a+i b x)^{\frac{i n}{2}}\left(1+\frac{1}{i a+i b x}\right)^{\frac{i n}{2}}}{(-1+i a+i b x)^{\frac{i n}{2}}}\,dx$$

$$\to\left(\frac{c}{1+a^2}\right)^p\frac{(i a+i b x)^{\frac{i n}{2}}\left(1+\frac{1}{i a+i b x}\right)^{\frac{i n}{2}}}{(1+i a+i b x)^{\frac{i n}{2}}}\frac{(1-i a-i b x)^{\frac{i n}{2}}}{(-1+i a+i b x)^{\frac{i n}{2}}}\int u\left(1+(a+b x)^2\right)^p\frac{(1+i a+i b x)^{\frac{i n}{2}}}{(1-i a-i b x)^{\frac{i n}{2}}}\,dx$$

$$\to\left(\frac{c}{1+a^2}\right)^p\left(\frac{i a+i b x}{1+i a+i b x}\right)^{\frac{i n}{2}}\left(\frac{1+i a+i b x}{i a+i b x}\right)^{\frac{i n}{2}}\frac{(1-i a-i b x)^{\frac{i n}{2}}}{(-1+i a+i b x)^{\frac{i n}{2}}}\int u(1-i a-i b x)^{p-\frac{i n}{2}}(1+i a+i b x)^{p+\frac{i n}{2}}\,dx$$

–

Program code:

```
Int[u_.*(c_+d_.*x_+e_.*x_^2)^p_.*E^(n_.*ArcCot[a_+b_.*x_]),x_Symbol] :=
  (c/(1+a^2))^p*((I*a+I*b*x)/(1+I*a+I*b*x))^(I*n/2)*((1+I*a+I*b*x)/(I*a+I*b*x))^(I*n/2)*
   ((1-I*a-I*b*x)^(I*n/2)/(-1+I*a+I*b*x)^(I*n/2))*
   Int[u*(1-I*a-I*b*x)^(p-I*n/2)*(1+I*a+I*b*x)^(p+I*n/2),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && Not[IntegerQ[I*n/2]] && EqQ[b*d-2*a*e,0] && EqQ[b^2*c-e(1+a^2),0] && (IntegerQ[p] || GtQ[c/(1+a^2),0])
```

2: $\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcCot}[a+b\,x]}\,dx$ when $\frac{i\,n}{2}\notin\mathbb{Z}\wedge b\,d==2\,a\,e\wedge b^2\,c-e\left(1+a^2\right)==0\wedge\neg\left(p\in\mathbb{Z}\vee\frac{c}{1+a^2}>0\right)$

Derivation: Piecewise constant extraction

Basis: If $b\,d==2\,a\,e\wedge b^2\,c-e\left(1+a^2\right)==0$, then $\partial_x\frac{\left(c+d\,x+e\,x^2\right)^p}{\left(1+a^2+2\,a\,b\,x+b^2\,x^2\right)^p}==0$

Rule: If $\frac{i\,n}{2}\notin\mathbb{Z}\wedge b\,d==2\,a\,e\wedge b^2\,c-e\left(1+a^2\right)==0\wedge\neg\left(p\in\mathbb{Z}\vee\frac{c}{1+a^2}>0\right)$, then

$$\int u\left(c+d\,x+e\,x^2\right)^p e^{n\,\text{ArcCot}[a+b\,x]}\,dx\to\frac{\left(c+d\,x+e\,x^2\right)^p}{\left(1+a^2+2\,a\,b\,x+b^2\,x^2\right)^p}\int u\left(1+a^2+2\,a\,b\,x+b^2\,x^2\right)^p e^{n\,\text{ArcCot}[a+b\,x]}\,dx$$

Program code:

```
Int[u_.*(c_+d_.*x_+e_.*x_^2)^p_.*E^(n_.*ArcCot[a_+b_.*x_]),x_Symbol] :=
  (c+d*x+e*x^2)^p/(1+a^2+2*a*b*x+b^2*x^2)^p*Int[u*(1+a^2+2*a*b*x+b^2*x^2)^p*E^(n*ArcCot[a*x]),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && Not[IntegerQ[I*n/2]] && EqQ[b*d-2*a*e,0] && EqQ[b^2*c-e(1+a^2),0] && Not[IntegerQ[p] || GtQ[c/(1+a^2),0]]
```

3: $\int u\, e^{n\,\text{ArcCot}\left[\frac{c}{a+b\,x}\right]}\,dx$

Derivation: Algebraic simplification

Basis: $\text{ArcCot}[z] == \text{ArcTan}\left[\frac{1}{z}\right]$

Rule:

$$\int u\, e^{n\,\text{ArcCot}\left[\frac{c}{a+b\,x}\right]}\,dx \longrightarrow \int u\, e^{n\,\text{ArcTan}\left[\frac{a}{c}+\frac{b\,x}{c}\right]}\,dx$$

Program code:

```
Int[u_.*E^(n_.*ArcCot[c_./(a_.+b_.*x_)]),x_Symbol] :=
  Int[u*E^(n*ArcTan[a/c+b*x/c]),x] /;
FreeQ[{a,b,c,n},x]
```